

Practicing Dynare 4.5.6

Francisco Barillas
New York University, Dept. of Economics

Anmol Bhandari
New York University, Dept. of Economics

Riccardo Colacito
University of North Carolina, Dept. of Finance

Sagiri Kitao
University of Southern California, FBE Department

Christian Matthes
New York University, Dept. of Economics

Thomas J. Sargent
New York University, Dept. of Economics

Yongseok Shin
Washington University, Dept. of Economics

Updated by:

Wenli Xu

Anhui University, School of Economics

CIMERS

March 10, 2019

Contents

1. Introduction	1
2. The Neoclassical growth model	1
2.1. The model	1
2.2. Calibration and approximation	2
2.2.1. Calibration	2
2.2.2. Approximation	3
2.3. Estimation	5
3. International Business Cycle Model	8
3.1. The model	8
3.2. Calibration and approximation	9
3.2.1. Calibration	9
3.3. Estimation	11
4. Fiscal policies in the growth model	14
4.1. The model	14
4.2. Parameter values	15
4.3. Transition experiments	15
4.3.1. A permanent increase in g	15
4.3.2. A permanent increase in τ_c	20
4.3.3. A permanent increase in τ_i	21
4.3.4. A permanent increase in τ_k	23
4.3.5. One time increase in g	24
4.3.6. One time increase in τ_i	26
5. Fiscal policies in the growth model	27
5.1. The model with inelastic labor supply	27
5.2. Parameter values	28
5.3. Transition experiments	29
5.3.1. Foreseen once-and-for-all increase in g	29
5.3.2. Foreseen once-and-for-all increase in τ_c	35
5.3.3. Foreseen once-and-for-all increase in τ_k	36
5.3.4. Foreseen one-time pulse increase in g	37
5.3.5. Foreseen once-and-for-all increase g at $t = 10$ for two economies	39
5.3.6. Impact of an foreseen once-and-for-all increase in g on yield curves	41

5.4. The model with elastic labor supply	44
5.4.1. Elastic Labor supply : Unforeseen once-and-for-all increase in g at t=0	45
5.4.2. Elastic Labor supply : Foreseen once-and-for-all increase in τ_n at t=0	50
5.4.3. Elastic Labor supply : Foreseen once-and-for-all increase in τ_n at t=10	51
6. The demand for money during hyperinflations	53
7. Solving and estimating Hall's permanent income model	58
7.1. The model	58
8. Solving and estimating Ryoo and Rosen's (2004) model of the market for engineers	61
8.1. The model	62
9. Estimation of Bansal and Yaron(2004) consumption growth process	67
10. Robust permanent income and pricing	71
11. Asset pricing with risks for the long run	76
11.1. Introduction	76
11.2. Setup of the model	76
11.3. Calibration and simulation	78
11.3.1. CRRA specification	78
11.3.2. Epstein-Zin specification	81
11.4. Bayesian estimation	83
12. References	87
A. Appendix List of Dynare Files	89

List of Figures

1.	Impulse response function to a technology shock	5
2.	Priors	7
3.	Posteriors	8
4.	Impulse response function(ϵ_1)	11
5.	Prior v. Posterior Distribution	13
6.	A permanent increase of g	20
7.	A permanent increase of τ_c	21
8.	A permanent increase of τ_i	23
9.	A permanent increase of τ_k	24
10.	One time increase of g	26
11.	One time increase of g	27
12.	Foreseen once-and-for-all increase in g at $t = 10$	34
13.	Foreseen once-and-for-all increase in τ_c at $t = 10$	36
14.	Foreseen once-and-for-all increase in τ_k at $t = 10$	37
15.	Foreseen one-time pulse increase in g at $t = 10$	38
16.	Response to foreseen once-and-for-all increase in g at $t = 10$	41
17.	Response to foreseen once-and-for-all increase in g at $t = 10$. Yield curves for $t = 0$ (solid line), $t = 10$ (dashdotted line) and $t = 60$ (dashed line); term to maturity is on the x axis for the yield curve, time for the other panels	44
18.	Elastic labor supply: response to unforeseen increase in g at $t = 0$. .	50
19.	Elastic labor supply: response to unforeseen increase in τ_n at $t = 0$.	51
20.	Elastic labor supply: response to unforeseen increase in τ_n at $t = 10$.	53
21.	the Log-Likelihood function for the hyperinflation model	56
22.	Posteriors and priors for the hyperinflation model	57
23.	A shock to ϵ_b	60
24.	priors and posteriors from the ML estimation of Hall's model	61
25.	A shock to $\epsilon_{s,t}$	64
26.	the Log-Likelihood for Ryoo and Rosen's model	65
27.	the Log-Likelihood for Ryoo and Rosen's model	66
28.	the Log-Likelihood function for the consumption growth model	69
29.	Prior and Posteriors for the Bansal and Yaron (2004) endowment process	70
30.	Prior and Posteriors for the Bansal and Yaron (2004) endowment process	71
31.	Impulse response function to a shock to x_t (CRRA preferences) . . .	81

32.	Impulse response function to a shock to x_t (Epstein-Zin preferences) .	83
33.	Posterior)	86

List of Tables

1.	parameter calibration	3
2.	Policy and transition functions	4
3.	Priors	7
4.	Posteriors	7
5.	parameter calibration	9
6.	parameter calibration	15
7.	parameter calibration	29
8.	Results from Maximum likelihood	56
9.	Results from Bayesian Estimation	66
10.	Results from Maximum likelihood	69
11.	Results from Bayesian Estimation	70
12.	Results from Bayesian Estimation	71
13.	parameter calibration.The value of θ is calibrated to satisfy equation (50).	79
14.	Policy and transition functions, CRRA calibration	81
15.	parameter calibration.The value of θ is calibrated to satisfy equation (50).	82
16.	Policy and transition functions, CRRA calibration	82
17.	Priors	85
18.	Priors and Posteriors	86

1. Introduction

This paper describes nine examples or types of examples that illustrate how Dynare can approximate the solutions of dynamic rational expectations models, simulate them, and estimate them by maximum likelihood and Bayesian methods. Section 2 approximates and estimates a one-sector stochastic growth model. Section 3 approximates and estimates a two-country stochastic growth model. Section 4 follows chapter 11 of Ljungqvist and Sargent (2004) in studying the effects of foreseen fiscal policy in a non-stochastic growth model. Section 5 updates and extends these examples to correspond to chapter 11 of Ljungqvist and Sargent (20XX). Section 6 estimates a rational expectations model of hyperinflation originally formulated by Sargent (1977). Section 7 solves and estimates the permanent income model of Hall (1988). This application is interesting, among other reasons, for the way that it illustrates how Dynare can implement the ‘diffuse Kalman filter’ needed in situations in which an initial endogenous state variable is unknown and the model has a unit root. Section 8 solves and estimates the Ryoo and Rosen (2004) rational expectations model of a market for engineers. Section 9 estimates a model of consumption growth proposed by Bansal and Yaron (2004). Section 10 solves and estimates the Hansen, Sargent, and Tallarini (1999) model of robust permanent income and pricing. Section 11 solves and estimates the Bansal and Yaron (2004) of asset prices and long-run risk in consumption and dividends. Appendix A tells how the reader can obtain the file examples.zip that contain the *.mod and data files that we used to generate these examples.

Some of our examples bear eloquent witness to the technological improvements brought by Dynare. For example, the maximum likelihood estimates in Sargent (1977) and Hansen, Sargent, and Tallarini (1999) were time-consuming and painful to obtain originally. Dynare has reduced the time and the pain.

2. The Neoclassical growth model

2.1. The model

We study a widely used stochastic neoclassical growth model with leisure (see, for example, Cooley and Prescott (1995)). A representative household’s problem is

$$\underbrace{\max}_{c_t, l_t} E_0 \sum_{t=1}^{\infty} \beta^{t-1} \frac{(c_t^{\theta} (1 - l_t)^{1-\theta})^{1-\tau}}{1 - \tau}$$

subject to the resource constraint

$$c_t + i_t = e^{z_t} k_t^\alpha l_t^{1-\alpha} \quad (1)$$

the law of motion for capital

$$k_{t+1} = i_t + (1 - \delta)k_t \quad (2)$$

and the stochastic process for productivity

$$z_t = \rho z_{t-1} + s_t \epsilon_t \quad (3)$$

with $\epsilon_t \sim N(0, \sigma^2)$.

The system of equations characterizing an equilibrium is comprised of equations (1), (2) and (3), the Euler intertemporal condition

$$\frac{c_t^\theta (1 - l_t)^{1-\theta})^{1-\tau}}{c_t} = \beta E_t \left[\frac{c_{t+1}^\theta (1 - l_{t+1})^{1-\theta})^{1-\tau}}{c_{t+1}} (1 + \alpha e^{z_t} k_t^{\alpha-1} l_t^\alpha - \delta) \right] \quad (4)$$

and an optimality condition for supply of labor

$$\frac{1 - \theta}{\theta} \frac{c_t}{1 - l_t} = (1 - \alpha) e^{z_t} k_t^\alpha l_t^{-\alpha} \quad (5)$$

Use (1) and (2) to get

$$k_{t+1} = e^{z_t} k_t^\alpha l_t^{1-\alpha} - c_t + (1 - \delta)k_t \quad (6)$$

An equilibrium is characterized by a system of four equations (3), (4), (5) and (6).

2.2. Calibration and approximation

2.2.1. Calibration

To estimate the policy function, parameters that enter the model are calibrated as in Table 1. For more on the calibrations of the related parameters, please see, for example, Cooley and Prescott (1995).

Table 1: parameter calibration

Parameter	Calibration
β	0.987
θ	0.357
δ	0.012
α	0.4
τ	2
ρ	0.95
s	0.007
σ	1

2.2.2. Approximation

The following instructions implement this model in Dynare.

```
var c k lab z;
```

```
varexo e;
```

```
parameters bet the del alp tau rho s;
```

```
bet = 0.987;
```

```
the = 0.357;
```

```
del = 0.012;
```

```
alp = 0.4;
```

```
tau = 2;
```

```
rho = 0.95;
```

```
s = 0.007;
```

```
model;
```

```
(c^the*(1-lab)^(1-the))^(1-tau)/c=bet*((c(+1)^(the*(1-lab(+1)))^(1-the))^(1-tau)/c(+1))* (1+alp*exp(z(-1))*k(-1)^(alp-1)*lab^(1-alp)-del);
```

```
c=the/(1-the)*(1-alp)*exp(z(-1))*k(-1)^(alp)*lab^(1-alp)*(1-lab);
```

```
k=exp(z(-1))*k(-1)^(alp)*lab^(1-alp)-c+(1-del)*k(-1);
```

```
z=rho*z(-1)+s*e;
```

```
end;
```

```
initval;
```

```

k = 1;
c = 1;
lab = 0.3;
z = 0;
e = 0;
end;

check;
resid;
steady;

shocks;
var e;
stderr 1;
end;

stoch_simul(periods=1000);

```

Dynare displays the results reported in Table 2 that provide the coefficients of the approximated policy functions and transition functions.

Table 2: Policy and transition functions

	c	k	lab	z
Constant	1.491628	29.288516	0.291593	0
(correction)	0.000002	-0.000004	0	0
k(-1)	0.028175	0.977868	-0.001880	0
z(-1)	0.629879	2.000368	0.207560	0.950000
e	0.002304	-0.004409	-0.000555	0.007000
k(-1),k(-1)	-0.000184	-0.000080	0.000026	0
z(-1),k(-1)	0.007775	0.025372	0.000613	0
z(-1),z(-1)	0.238361	1.341812	-0.004369	0
e,e	0.000005	-0.000008	0	0
k(-1),e	0.000024	-0.000037	0.000006	0
z(-1),e	0.001509	-0.003929	-0.000241	0

In fact, by default Dynare solves models using a second order approximation. For a 2nd order approximation, Dynare will produce the following output for the state space representation:

$$s_t = \frac{1}{2}\Gamma_{s,0} + As_{t-1} + B\epsilon_t + \frac{1}{2}\Gamma_{s,1}(s_{t-1} \otimes_{t-1}) + \frac{1}{2}\Gamma_{s,2}(\epsilon_t \otimes \epsilon_t) + \Gamma_{s,3}(s_{t-1} \otimes \epsilon_t)$$

$$x_t = \frac{1}{2}\Gamma_{x,0} + Cs_{t-1} + D\epsilon_t + \frac{1}{2}\Gamma_{x,1}(s_{t-1} \otimes_{t-1}) + \frac{1}{2}\Gamma_{x,2}(\epsilon_t \otimes \epsilon_t) + \Gamma_{x,3}(s_{t-1} \otimes \epsilon_t)$$

In table 2, what Dynare labels "correction" is the coefficient $\Gamma_{s,0}$ and $\Gamma_{x,0}$. The coefficients on $k(-1), z(-1), e$ are the values in the A, B, C, D matrixes above. The coefficients on $k(-1), k(-1), z(-1), k(-1), z(-1), z(-1)$ are the elements $\Gamma_{s,1}, \Gamma_{x,1}$. The coefficients on e, e are the $\Gamma_{s,2}, \Gamma_{x,2}$ matrixes, while the $k(-1), e$ and $z(-1), e$ coefficients are the elements of $\Gamma_{s,3}, \Gamma_{x,3}$.

Dynare also displays the result reported in Fig 1 that provide the impulse response function to a technology shock.




Figure 1: Impulse response function to a technology shock

2.3. Estimation

The simulated data series for consumption are used to estimate some unknown parameters of the model. Suppose we would like to estimate the preference parameters θ and τ , and the stochastic process for productivity, summarized by two parameters ρ and σ . The following code instructs Dynare to conduct the estimation.

```
var c k lab z;
varexo e;
```

```
parameters bet del alp rho the tau s;
```

```
bet = 0.987;
```

```
the = 0.357;
```

```
del = 0.012;
```

```
alp = 0.4;
```

```
tau = 2;
```

```
rho = 0.95;
```

```
s = 0.007;
```

```
model;
```

```
(c^the*(1-lab)^(1-the))^(1-tau)/c=bet*((c(+1)^the*(1-lab(+1))^(1-the))^(1-tau)/c(+1))*(1+alp*exp(1)*lab(+1)^(1-alp)-del);
```

```
c=the/(1-the)*(1-alp)*exp(z)*k(-1)^alp*lab^(-alp)*(1-lab);
```

```
k=exp(z)*k(-1)^alp*lab^(1-alp)-c+(1-del)*k(-1);
```

```
z=rho*z(-1)+s*e;
```

```
end;
```

```
initval;
```

```
k = 1;
```

```
c = 1;
```

```
lab = 0.3;
```

```
z = 0;
```

```
e = 0;
```

```
end;
```

```
shocks;
```

```
var e;
```

```
stderr 1;
```

```
end;
```

```
estimated_params;
```

```
stderr e, inv_gamma_pdf, 0.95,30;
```

```
rho, beta_pdf,0.93,0.02;
```

```
the, normal_pdf,0.3,0.05;
```

```
tau, normal_pdf,2.1,0.3;
```

```
end;
```

varobs c;

estimation(datafile=simuldata,mode__compute=6,mh_replic=1000,mh_jscale=0.9,nodiagnostic);

The priors used in estimation are as in Table 3. Table 4 is one of the outputs of Dynare which presents summary statistics of posterior distribution.

Table 3: Priors

Parameter	Distribution	Mean	Std.Dev.
ρ	Beta	0.93	0.02
θ	Normal	0.3	0.05
τ	Normal	2.1	0.3
σ	Inv.Gamma	0.95	30

Table 4: Posteriors

Parameter	prior mean	post.mean	conf.	interval	prior dist.	prior Std.
ρ	0.9300	0.7451	0.7383	0.7530	beta	0.0200
θ	0.300	0.3542	0.3500	0.3582	norm	0.0500
τ	2.100	0.7613	0.7542	0.7689	norm	0.3000
σ	0.9500	4.4338	4.3725	4.5074	invg	30.00




Figure 2: Priors




Figure 3: Posteriors

3. International Business Cycle Model

We study a simplified version of the two-country production model in Kim and Kim (2003).

3.1. The model

Two countries are identical ex-ante and markets are complete. A solution to a Pareto planner's problem is characterized by the following equations

$$\begin{aligned}
 C_{1,t} &= C_{2,t} \\
 C_{1,t}^{-\gamma} &= \beta E_t C_{1,t}^{-\gamma} (\alpha A_{1,t+1} K_{1,t+1}^{\alpha-1} + 1 - \delta) \\
 C_{2,t}^{-\gamma} &= \beta E_t C_{2,t}^{-\gamma} (\alpha A_{2,t+1} K_{2,t+1}^{\alpha-1} + 1 - \delta) \\
 A_{1,t} K_{1,t}^\alpha + A_{2,t} K_{2,t}^\alpha &= C_{1,t} + C_{2,t} + K_{1,t-1} - (1 - \delta) K_{1,t} + K_{2,t-1} - (1 - \delta) K_{2,t} \\
 \ln A_{1,t+1} &= \rho \ln A_{1,t} + \epsilon_{1,t+1} \\
 \ln A_{2,t+1} &= \rho \ln A_{2,t} + \epsilon_{2,t+1}
 \end{aligned}$$

The first three equalities are the first-order conditions with regard to the consumption ($C_{i,t}$), and the fourth one is the world resource constraint. The last two describe the law of motion for the technological shock ($A_{i,t}$). $K_{i,t}$ denotes the capital stock of

country i at time t .

3.2. Calibration and approximation

3.2.1. Calibration

The following parameter values are used.

Table 5: parameter calibration

Parameter	Calibration
β	0.98
δ	0.05
α	0.4
ρ	0.85
σ_{ϵ_1}	0.08
σ_{ϵ_2}	0.08

Dynare computes the policy function as a second-order approximation around the (log) steady-state. The .mod Dynare file of the system is as follows.

```
var c1 c2 k1 k2 a1 a2;
```

```
varexo e1 e2;
```

```
parameters gamma delta alpha beta rho;
```

```
gamma=2;
```

```
delta=.05;
```

```
alpha=.4;
```

```
beta=.98;
```

```
rho=.85;
```

```
model;
```

```
c1=c2;
```

```
exp(c1)^(-gamma) = beta*exp(c1(+1))^(gamma)*(alpha*exp(a1(+1)))*exp(k1)^(alpha-1)+1-delta);
```

```
exp(c2)^(-gamma) = beta*exp(c2(+1))^(gamma)*(alpha*exp(a2(+1)))*exp(k2)^(alpha-1)+1-delta);
```

```
exp(c1)+exp(c2)+exp(k1)-exp(k1(-1))*(1-delta)+exp(k2)-exp(k2(-1))*(1-delta)= exp(a1)*exp(k1(-1))^alpha+exp(a2)*exp(k2(-1))^alpha;
```

```
a1=rho*a1(-1)+e1;
```

```

a2=rho*a2(-1)+e2;
end;

initval;
k1=2.8;
k2=2.8;
c1=.8;
c2=.8;
a1=0;
a2=0;
e1=0;
e2=0;
end;

shocks;
var e1;
stderr .08;
var e2;
stderr .08;
end;

steady;

stoch_simul(periods=1000);

```

The output is given as a matrix of coefficients for the policy function. From this matrix, for example, one can construct the optimal consumption rule for Country 1:

$$\begin{aligned}
\hat{c}_{1,t} = & 0.004 + 0.243(\hat{k}_{1,t-1} + \hat{k}_{2,t-1}) + 0.119(\hat{a}_{1,t-1} + \hat{a}_{2,t-1}) \\
& + 0.140(\hat{e}_{1,t} + \hat{e}_{2,t}) + 0.062(\hat{k}_{1,t-1}^2 + \hat{k}_{2,t-1}^2) \\
& - 0.110\hat{k}_{1,t-1}\hat{k}_{2,t-1} - 0.013(\hat{k}_{1,t-1}\hat{a}_{1,t-1} + \hat{k}_{2,t-1}\hat{a}_{2,t-1}) \\
& - 0.027(\hat{k}_{2,t-1}\hat{a}_{1,t-1} + \hat{k}_{1,t-1}\hat{a}_{2,t-1}) + 0.032(\hat{a}_{1,t-1}^2 + \hat{a}_{2,t-1}^2) \\
& - 0.029\hat{a}_{1,t-1}\hat{a}_{2,t-1} + 0.044(\hat{e}_{1,t}^2 + \hat{e}_{2,t}^2) - 0.040\hat{e}_{1,t}\hat{e}_{2,t} \\
& - 0.015(\hat{k}_{1,t-1}\hat{e}_{1,t} + \hat{k}_{2,t-1}\hat{e}_{2,t}) + 0.075(\hat{a}_{1,t-1}\hat{e}_{1,t} + \hat{a}_{2,t-1}\hat{e}_{2,t}) \\
& - 0.34(\hat{a}_{1,t-1}\hat{e}_{2,t} + \hat{a}_{2,t-1}\hat{e}_{1,t})
\end{aligned}$$

where \hat{c} denotes the deviation of $\ln C$ from its log steady-state. Note that Dynare breaks $\hat{a}_{i,t}$ into $\hat{a}_{i,t-1}$ and $\epsilon_{i,t}$, so the output from Dynare looks unnecessarily messy.

Dynare also produces the associated impulse response functions of the system. Figure 4 illustrate the response of the system to a one-standard-deviation shock to ϵ_1 .




Figure 4: Impulse response function(ϵ_1)

The last line of the code generates sample paths of the variables governed by the approximated policy functions.

3.3. Estimation

Now assume that we do not know the true parameter values. We can estimate the unknown parameters from the saved data, using the Bayesian econometrics function of Dynare.

We will estimate 4 parameters $\rho, \alpha, \sigma_{\epsilon_1}$ and σ_{ϵ_2} . Hence, we have to specify the prior distribution of each parameter. In this exercise, ρ, α are given a normal prior, while the priors of σ_{ϵ_1} and σ_{ϵ_2} are inverse gamma distributions. The Dynare code is as follows.

```
var c1 c2 k1 k2 a1 a2;
varexo e1 e2;
```

```

parameters gamma delta alpha beta rho;

gamma=2;
delta=.05;
alpha=.4;
beta=.98;
rho=.85;

model;
c1=c2; exp(c1)^(1-gamma)=beta*exp(c1+1))^(1-gamma)*(alpha*exp(a1+1))*exp(k1)^(alpha-
1)+1-delta);
exp(c2)^(1-gamma)=beta*exp(c2+1))^(1-gamma)*(alpha*exp(a2+1))*exp(k2)^(alpha-1)+1-
delta);
exp(c1)+exp(c2)+exp(k1)-exp(k1(-1))*(1-delta)+exp(k2)-exp(k2(-1))*(1-delta)=exp(a1)*exp(k1(-
1))^alpha+exp(a2)*exp(k2(-1))^alpha;
a1=rho*a1(-1)+e1;
a2=rho*a2(-1)+e2;
end;

initval;
k1=2.8;
k2=2.8;
c1=.8;
c2=.8;
a1=0;
a2=0;
e1=0;
e2=0;
end;

shocks;
var e1;
stderr .08;
var e2;
stderr .08;
end;

```

```
steady;
```

```
estimated_params;
```

```
rho, normal_pdf, .84,.05;
```

```
alpha, normal_pdf, .38, .03;
```

```
stderr e1, inv_gamma_pdf, .078, inf;
```

```
stderr e2,inv_gamma_pdf,.082, inf;
```

```
end;
```

```
varobs c1 k2;
```

```
estimation(datafile=simuldata_international,mode_compute=6,mh_replic=1000,mh_jscale=.5);
```

Given that there are two exogenous shock variables, we use two observables $\ln C_1$ and $\ln K_2$. The result of the estimation is summarized in Figure 5, where the posterior distribution (in the darker line) of each parameter is contrasted against the given prior distribution. The result can be compared to the true parameter values in Table 5.




Figure 5: Prior v. Posterior Distribution

4. Fiscal policies in the growth model

This section describes a suite of Dynare programs that replicate the transition experiments performed in chapter 11 of Ljungqvist and Sargent (2004).

4.1. The model

A deterministic growth model has inelastic labor supply, an exogenous stream of government expenditures, and several kinds of distorting taxes. All variables are as described in chapter 11 of Ljungqvist and Sargent (2004). A representative agent maximizes

$$\sum_{t=0}^{\infty} \beta^t \frac{c_t^{1-\gamma}}{1-\gamma}, \quad \beta \in (0, 1)$$

Feasible allocations satisfy

$$g_t + c_t + k_{t+1} \leq A_t K_t^\alpha + (1 - \text{delta})k_t$$

The household budget constraint is

$$\sum_{t=0}^{\infty} \{q_t(1 + \tau_{c,t}) + (1 - \tau_{i,t})q_t[k_{t+1} - (1 - \delta)k_t]\} \leq \sum_{t=0}^{\infty} \{r_t(1 - \tau_{k,t})k_t + w_t\}$$

and the government budget constraint is

$$\sum_{t=0}^{\infty} q_t g_t \leq \sum_{t=0}^{\infty} \{\tau_{c,t} q_t c_t - \tau_{i,t} q_t [k_{t+1} - (1 - \delta)k_t] + r_t \tau_{k,t} k_t + w_t\}$$

The following conditions characterize an equilibrium:

$$c_t = A_t k_t^\alpha + (1 - \delta)k_t - k_{t+1} - g_t \quad (7)$$

$$q_t = \beta^t \frac{c_t^{1-\gamma}}{1 + \tau_{c,t}} \quad (8)$$

$$\frac{r_t}{q_t} = A_t \alpha k_t^{\alpha-1} \quad (9)$$

$$\frac{w_t}{q_t} = A_t k_t^\alpha - k_t A_t \alpha k_t^{\alpha-1} \quad (10)$$

$$R_{t+1} = \frac{1 + \tau_{c,t}}{1 + \tau_{c,t+1}} \left[\frac{1 + \tau_{i,t+1}}{1 + \tau_{i,t}} (1 - \delta) + \frac{1 + \tau_{k,t+1}}{1 + \tau_{k,t}} A_t h = k_{t+1}^{\alpha-1} \right] \quad (11)$$

$$\frac{s_t}{q_t} = (1 - \tau_{k,t}) A_t \alpha k_t^{\alpha-1} + (1 - \delta) \quad (12)$$

$$c_t^{-\gamma} = \beta c_{t+1}^{-\gamma} R_{t+1} \quad (13)$$

Because the other endogenous variables can be expressed as functions of k_t and c_t , in the programs below we use only a subset of the equilibrium conditions to compute equilibrium paths.

4.2. Parameter values

We set the parameters and the baseline values of the exogenous variables at values set in chapter 11 of Ljungqvist and Sargent (2004), as described in the following table 6 are taken from RMT2.

Table 6: parameter calibration

Parameter	Calibration
β	0.95
δ	0.2
A	1
α	0.33
γ	2
g	0.2
τ_c	0
τ_i	0
τ_k	0

4.3. Transition experiments

4.3.1. A permanent increase in g

The Dynare code below reproduces Figure 6. which looks at the impact of a permanent increase of 0.2 in g at $t=10$.

```
% This program replicates figure 11.3.1 from chapter 11 of RMT2 by
% Ljungqvist and Sargent.
% y_ records the simulated endogenous variables in alphabetical order
% ys0_ records the initial steady state
% ys_ records the terminal steady state
% We check that these line up at the end points
% Note: y_ has ys0_ in the first column, ys_ in the last column, which explains
% why it has 102 elements.
% The sample of size 100 is in between.
```

```

% Warning: All endogenous variables dated time t are treated as jump variables
% in Dynare. So k in the program corresponds to k_t+1
% and the same timing holds for the taxes.
% Declares the endogenous variables;

var c k;

% declares the exogenous variables
% investment tax credit, consumption tax, capital tax, government spending

varexo tau_i tau_c tau_k g;

parameters bet gam del alpha A;

bet=.95; % discount factor
gam=2; % CRRA parameter
del=.2; % depreciation rate
alpha=.33; % capital's share
A=1; % productivity

% Alignment convention:
% g tau_c tau_i tau_k are now columns of ex_. Because of a bad design decision
% the date of ex_(1,:) doesn't necessarily match the date in y_.
% Whether they match depends on the number of lag periods in endogenous versus
% exogenous variables.
% In this example they match because tau_c(-1) and tau_i(-1) enter the model.
% These decisions and the timing conventions mean that y_(:,1) records the
% initial steady state, while y_(:,102) records the terminal steady state
% values. For j > 2, y_(:,j) records [c(j-1) .. k(j-1) .. G(j-1)] where k(j-1)
% means end of period capital in period j-1, which equals k(j) in chapter 11
% notation. Note that the jump in G occurs in y_(:,11), which confirms this
% timing. The jump occurs now in ex_(11,1)

model;
% equation 11.3.8.a
k=A*k(-1)^alpha+(1-del)*k(-1)-c-g;
% equation 11.3.8e + 11.3.8.g

```

```

c(-gam)=bet*(c(+1)^(1-gam))*((1+tauc(-1))/(1+tauc))*((1-taii)*(1-del)/(1-taii(-1)))+((1-
tauk)/(1-taii(-1)))*alpha*A*k(-1)^(alpha-1));
end;

initval;
k=1.5;
c=0.6;
g = 0.2;
tauc = 0;
taii = 0;
tauk = 0;
end;

steady; % put this in if you want to start from the initial
% steady state, comment it out to start from the indicated values

endval; % The following values determine the new steady state
% after the shocks.
k=1.5;
c=0.4;
g =.4;
tauc =0;
taii =0;
tauk =0;
end;

steady; % We use steady again and the enval provided are initial % guesses for
Dynare to compute the ss.
% The following lines produce a g sequence with a once and for all jump in g shocks
% we use shocks to undo that for the first 9 periods and leave g at
% it's initial value of 0.2

shocks;
var g;
periods 1:9;
values 0.2;
end;

```

```

% now solve the model

simul(periods=100);

% Note: y_ records the simulated endogenous variables in order of declaration
% ys0_ records the initial steady state
% ys_ records the terminal steady state
% check that these line up at the end points
load('neoclassical_fiscal_results.mat', 'oo_');
y_=oo_.endo_simul;
ys0_(:)=y_(:,1);
ys_(:)=y_(:,102);

% Compute the initial steady state for consumption to later do the plots.
c0=y_(1,1);
k0 = y_(2,1);
% g is in oo_.exo_simul(:,4) since it is stored in order of declaration
g0 = oo_.exo_simul(1,4);

% The following equation compute the other endogenous variables use in the plots
below
% Since they are function of capital and consumption, so we can compute them from
the solved
% model above.

%
rbig0=1/bet;
rbig=y_(1,2:101).^(1-gam)./(bet*y_(1,3:102).^(1-gam));
rq0=alpha*A*k0^(alpha-1);
rq=alpha*A*y_(2,1:100).^(alpha-1);
wq0=A*k0^alpha-k0*alpha*A*k0^(alpha-1);
wq=A*y_(2,1:100).^alpha-y_(2,1:100).*alpha*A.*y_(2,1:100).^(alpha-1);
sq0=(1-oo_.exo_simul(1,3))*A*alpha*k0^(alpha-1)+(1-del);
sq=(1-oo_.exo_simul(1:100,3))'*A*alpha.*y_(2,1:100).^(alpha-1)+(1-del);

%Now we plot the responses of the endogenous variables to the shock.

```



```

figure
subplot(2,3,1)
plot([k0*ones(100,1) y_(2,1:100)'] ) % note the timing: we lag capital to correct for
syntax
title('k')
subplot(2,3,2)
plot([c0*ones(100,1) y_(1,2:101)'] )
title('c')
subplot(2,3,3)
plot([rbig0*ones(100,1) rbig' ])
title('R')
subplot(2,3,4)
plot([wq0*ones(100,1) wq' ])
title('w/q')
subplot(2,3,5)
plot([sq0*ones(100,1) sq' ])
title('s/q')
subplot(2,3,6)
plot([rq0*ones(100,1) rq' ])
title('r/q')

```

The following figure 6 shows the path of the endogenous variables along the transition to the new steady state.




Figure 6: A permanent increase of g

Next we show the other transition experiments and only the parts of the code that need to be changed for each experiment.

4.3.2. A permanent increase in τ_c

A permanent increase of τ_c at $t=10$ of 20 percent(Figure 7).

```
initval;
k=1.5;
c=0.6;
g = 0.2;
tauc = 0;
taui = 0;
tauk = 0;
end;
```

```
steady;
```

```
endval;
k=1.5;
c=0.6;
```

```

g = 0.2;
tauc =0.2;
taui =0;
tauk =0;
end;
shocks;
var tauc;
periods 1:9;
values 0;
end;

```




Figure 7: A permanent increase of τ_c

4.3.3. A permanent increase in τ_i

A permanent increase of τ_i at $t=10$ of 20 percent(Figure 8).

```

initval;
k=1.5;
c=0.6;
g = 0.2;
tauc = 0;

```

```

    tau_i = 0;
    tau_k = 0;
end;

steady;

endval;
k=1.5;
c=0.6;
g =0.2;
tau_c =0;
tau_i =0.20;
tau_k =0;
end;

steady;

shocks;
var tau_i;
periods 1:9;
values 0;
end;

```




Figure 8: A permanent increase of τ_i

4.3.4. A permanent increase in τ_k

A permanent increase of τ_k at $t=10$ of 20 percent(Figure 9).

```

initval;
k=1.5;
c=0.6;
g = 0.2;
tauc = 0;
taui = 0;
tauk = 0;
end;

```

```

steady;

```

```

endval;
k=1.5;
c=0.6;
g =0.2;
tauc =0;
taui =0;

```

```

tauk = 0.2;
end;

steady;

shocks;
var tauk;
periods 1:9;
values 0;
end;

```




Figure 9: A permanent increase of τ_k

4.3.5. One time increase in g

A one time pulse of g at $t=10$ of 0.2. (Figure 10)

```

initval;
k=1.5;
c=0.6;
g = 0.2;
tauc = 0;

```

```
    tau_i = 0;  
    tau_k = 0;  
end;
```

```
steady;
```

```
endval;  
k=1.5;  
c=0.6;  
g = 0.2;  
tau_c =0;  
tau_i =0;  
tau_k =0;  
end;
```

```
steady;
```

```
shocks;  
var g;  
periods 10;  
values 0.4;  
end;
```




Figure 10: One time increase of g

4.3.6. One time increase in τ_i

A one time pulse of τ_i at $t=10$ of 20 percent. (Figure 11)

```

initval;
k=1.5;
c=0.6;
g = 0.2;
tauc = 0;
taui = 0;
tauk = 0;
end;

```

```

steady;

```

```

endval;
k=1.5;
c=0.6;
g = 0.2;
tauc = 0;
taui = 0;

```



```

tau_k = 0;
end;

steady;

shocks;
var tau_i;
periods 10;
values 0.2;
end;

```




Figure 11: One time increase of g

5. Fiscal policies in the growth model

This section describes a suite of Dynare programs that replicate the transition experiments performed in chapter 11 of Ljungqvist and Sargent (2004).

5.1. The model with inelastic labor supply

A deterministic growth model has inelastic labor supply, an exogenous stream of government expenditures, and several kinds of distorting taxes. All variables are as

described in chapter 11 of Ljungqvist and Sargent (2004). A representative agent maximizes

$$\sum_{t=0}^{\infty} \beta^t \frac{c_t^{1-\gamma}}{1-\gamma}, \quad \beta \in (0, 1)$$

Feasible allocations satisfy

$$g_t + c_t + k_{t+1} \leq A_t K_t^\alpha + (1 - \delta)k_t$$

The household budget constraint is

$$\sum_{t=0}^{\infty} \{q_t c_t (1 + \tau_{c,t}) + q_t [k_{t+1} - (1 - \delta)k_t]\} \leq \sum_{t=0}^{\infty} q_t \{\eta_t (1 - \tau_{k,t}) k_t + w_t\}$$

and the government budget constraint is

$$\sum_{t=0}^{\infty} q_t g_t \leq \sum_{t=0}^{\infty} q_t \{\tau_{c,t} c_t + \eta_t \tau_{k,t} k_t\}$$

The following conditions characterize an equilibrium:

$$c_t = A_t k_t^\alpha + (1 - \delta)k_t - k_{t+1} - g_t \quad (14)$$

$$q_t = \beta^t \frac{c_t^{-\gamma}}{1 + \tau_{c,t}} \quad (15)$$

$$\eta_t = A_t \alpha k_t^{\alpha-1} (\alpha - 1) \quad (16)$$

$$w_t = A_t k_t^\alpha (\alpha - 1) - k_t A_t \alpha k_t^{\alpha-1} (\alpha - 1) \quad (17)$$

$$\hat{R}_{t+1} = \frac{1 + \tau_{c,t}}{1 + \tau_{c,t+1}} [(1 - \delta) + (1 - \tau_{k,t+1}) A_t \alpha k_{t+1}^{\alpha-1}] \quad (18)$$

$$c_t^{-\gamma} = \beta c_{t+1}^{-\gamma} \hat{R}_{t+1} \quad (19)$$

Because the other endogenous variables can be expressed as functions of k_t and c_t , in the programs below we use only a subset of the equilibrium conditions to compute equilibrium paths.

5.2. Parameter values

We set the parameters and the baseline values of the exogenous variables at values set in chapter 11 of Ljungqvist and Sargent (2004), as described in the following table 6 are taken from RMT2.

Table 7: parameter calibration

Parameter	Calibration
β	0.95
δ	0.2
A	1
α	0.33
γ	2
g	0.2
τ_c	0
τ_k	0

5.3. Transition experiments

5.3.1. Foreseen once-and-for-all increase in g

The Dynare code below reproduces Figure 12. which looks at the impact of a foreseen once-for-and-all increase of 0.2 in g at $t = 10$.

```
%-----  
% Experiment : Permenant increase in g at t=10  
% Note:Following the discussion in the text  $t = 0$  is the first period of the new  
(simulated) path  
%-----  
  
% This program replicates figure 11.6.1 from chapter 11 of RMT3 by Ljungqvist and  
Sargent  
  
% Dynare records the endogenous variables with the following convention.  
% Say  $N$  is the number of simulations(sample)  
% Index 1 : Initial values (steady sate)  
% Index 2 to  $N+1$  :  $N$  simulated values  
% Index  $N+2$  : Terminal Value (Steady State)  
% Warning: we align  $c$ ,  $k$ , and the taxes to exploit the dynare syntax.  
% In Dynare the timing of the variable reflects the date when the variable is decided.  
% For instance the capital stock for time ' $t$ ' is decided in time ' $t-1$ '(end of period).  
% So a statement like  $k(t+1) = i(t) + (1-\text{del})*k(t)$  would translate to  
%  $k(t) = i(t) + (1-\text{del})*k(t-1)$  in the code.  
  
%-----
```

```
% 1. Defining variables
```

```
%
```

```
% Declares the endogenous variables consumption ('c') capital stock (k);
```

```
var c k;
```

```
% declares the exogenous variables consumption tax ('tauc'),
```

```
% capital tax('tauk'), government spending('g')
```

```
varexo tauc tauk g;
```

```
parameters bet gam del alpha A;
```

```
%
```

```
% 2. Calibration and alignment convention
```

```
%
```

```
bet=.95; % discount factor
```

```
gam=2; % CRRA parameter
```

```
del=.2; % depreciation rate
```

```
alpha=.33; % capital's share
```

```
A=1; % productivity
```

```
% Alignment convention:
```

```
% g tauc tauk are now columns of ex_. Because of a bad design decision the date of  
ex_(1,:)
```

```
% doesn't necessarily match the date in endogenous variables. Whether they match  
depends on
```

```
% the number of lag periods in endogenous versus exogenous variables.
```

```
% These decisions and the timing conventions mean that
```

```
% k(1) records the initial steady state, while k(102) records the terminal steady state  
values.
```

```
% For  $j > 1$ , k(j) records the variables for j-1th simulation where the capital stock  
decision
```

```
% taken in j-1th simulation i.e stock at the beginning of period j.
```

```
% The variable ex_ also follows a different timing convention i.e
```

```
% ex_(j,:) records the value of exogenous variables in the jth simulation.
```

```

% The jump in the government policy is reflected in ex__(11,1) for instance.

%-----
% 3. Model
%-----

model;
% equation 11.3.8.a

k=A*k(-1)^alpha+(1-del)*k(-1)-c-g;

% equation 11.3.8e + 11.3.8.g
c^(1-gam)=bet*(c(+1)^(1-gam))*((1+tauc)/(1+tauc(+1)))*((1-del)+(1-tauk(+1))^alpha*A*k^(alpha-1));

end;

%-----
% 4. Computation
%-----

initval;
k=1.5;
c=0.6;
g = .2;
tauc = 0;
tauk = 0;
end;

steady;
% put this in if you want to start from the initial steady state,comment it out to
start
% from the indicated values

% The following values determine the new steady state after the shocks.
endval;
k=1.5;

```

```

c=0.6;
g =.4;
tauc =0;
tauk =0;
end;

steady;
% We use 'steady' again and the endval provided are initial guesses for dynare to
compute the ss.

% The following lines produce a g sequence with a once and for all jump in g
% we use 'shocks' to undo that for the first 10 periods (t=0 until t=9)and leave g at
% it's initial value of 0
% Note : period j refers to the value in the jth simulation

shocks;
var g;
periods 1:10;
values 0.2;
end;

% now solve the model
simul(periods=100);

% Compute the initial steady state for consumption to later do the plots.
load('neoclassical_fiscal_fig12_results.mat', 'oo_');
c0=oo_.endo_simul(1,1);
k0 =oo_.endo_simul(2,1);
% g is in oo_.exo_simul(:,1) since it is stored in order of declaration
g0 = oo_.exo_simul(1,3)

%-----
% 5. Graphs and plots for other endogenous variables
%-----

% Let N be the periods to plot
N=40;

```

```

% The following equation compute the other endogenous variables use in the plots
below
% Since they are function of capital and consumption, so we can compute them from
the solved
% model above.
% These equations were taken from page 371 of RMT3
rbig0=1/bet;
rbig=oo_.endo_simul(1,2:101).^(1-gam)./(bet*oo_.endo_simul(1,3:102).^(1-gam));
nq0=alpha*A*k0^(alpha-1);
nq=alpha*A*oo_.endo_simul(2,1:100).^(alpha-1);
wq0=A*k0^alpha-k0*alpha*A*k0^(alpha-1);
wq=A*oo_.endo_simul(2,1:100).^alpha-oo_.endo_simul(2,1:100).*alpha*A.*oo_.endo_simul(2,1:100)
1);

% Now we plot the responses of the endogenous variables to the shock.
x=0:N-1;
figure(1)

% subplot for capital 'k'
subplot(2,3,1)
plot(x,[k0*ones(N,1)],'-k', x,oo_.endo_simul(2,1:N),'k','LineWidth',1.5)
% note the timing: we lag capital to correct for syntax
title('k','FontSize',12)
set(gca,'FontSize',12)

% subplot for consumption 'c'
subplot(2,3,2)
plot(x,[c0*ones(N,1)],'-k', x,oo_.endo_simul(1,2:N+1),'k','LineWidth',1.5)
title('c','FontSize',12)
set(gca,'FontSize',12)

% subplot for cost of capital 'R_bar'
subplot(2,3,3)
plot(x,[rbig0*ones(N,1)],'-k', x,rbig(1:N),'k','LineWidth',1.5)
title('\overline{R}','interpreter','latex','FontSize',12)
set(gca,'FontSize',12)

```

```

% subplot for rental rate 'eta'
subplot(2,3,4)
plot(x,[nq0*ones(N,1)],'-k', x,nq(1:N),'k','LineWidth',1.5)
title('\eta','FontSize',12)
set(gca,'FontSize',12)

% subplot for the experiment proposed
subplot(2,3,5)
plot([0:9],oo_.exo_simul(1:10,3),'k','LineWidth',1.5);
hold on;
plot([10:N-1],oo_.exo_simul(11:N,3),'k','LineWidth',1.5);
hold on;
plot(x,[g0*ones(N,1)],'-k','LineWidth',1.5)
title('g','FontSize',12)
axis([0 N -.1 .5])
set(gca,'FontSize',12)

```

The following figure shows the path of the endogenous variables along the transition to the new steady state.




Figure 12: Foreseen once-and-for-all increase in g at $t = 10$

Next we show the other transition experiments and only the parts of the code that

need to be changed for each experiment.

5.3.2. Foreseen once-and-for-all increase in τ_c

A foreseen once-and-for-all increase in τ_c at $t = 10$ of 20 per cent.(Figure 13)

```
initval;  
k=1.5;  
c=0.6;  
g = 0.2;  
tauc = 0;  
tauk = 0;  
end;  
  
steady;  
  
endval;  
k=1.5;  
c=0.6;  
g = 0.2;  
tauc =0.2;  
tauk =0;  
end;  
  
steady;  
  
shocks;  
var tauc;  
periods 1:10;  
values 0;  
end;
```




Figure 13: Foreseen once-and-for-all increase in τ_c at $t = 10$

5.3.3. Foreseen once-and-for-all increase in τ_k

A foreseen once-and-for-all increase in τ_k at $t = 10$ of 20 per cent.(Figure 14)

```

initval;
k=1.5;
c=0.6;
g = 0.2;
tauc = 0;
tauk = 0;
end;

steady;

endval;
k=1.5;
c=0.6;
g =0.2;
tauc =0;
tauk = 0.2
; end;

```

```

steady;

shocks;
var tauk;
periods 1:10;
values 0;
end;

```




Figure 14: Foreseen once-and-for-all increase in τ_k at $t = 10$

5.3.4. Foreseen one-time pulse increase in g

A foreseen one-time pulse of g at $t = 10$ of 0.2. (Figure 15)

```

initval;
k=1.5;
c=0.6;
g = 0.2;
tauc = 0;
tauk = 0;
end;

```

```
steady;
```

```
endval;
```

```
k=1.5;
```

```
c=0.6;
```

```
g = 0.2;
```

```
tauc =0;
```

```
tauk =0;
```

```
end;
```

```
steady;
```

```
shocks;
```

```
var g;
```

```
periods 11;
```

```
values 0.4;
```

```
end;
```




Figure 15: Foreseen one-time pulse increase in g at $t = 10$

5.3.5. Foreseen once-and-for-all increase g at $t = 10$ for two economies

Responses to a foreseen increase in g at $t = 10$ for two economies, the original economy with $\gamma = 2$ and an otherwise identical economy with $\gamma = 0.2$ (Figure 16).

```
textcolorblue % Compute the initial steady state for consumption to later do the plots.
```

```
c0=c(1);
```

```
k0= k(1);
```

```
% g is in oo_.exo_simul(:,1) since it is stored in order of declaration
```

```
g0 = oo_.exo_simul(1,1);
```

```
%values for gamma = 2
```

```
kgamma(:,1)=k;
```

```
cgamma(:,1)=c;
```

```
rbig(:,1)=cgamma(2:101,1). $\hat{(-gam)}$ ./(bet*cgamma(3:102,1). $\hat{(-gam)}$ ));
```

```
%values for gamma=.2
```

```
gam=.2;
```

```
simul(periods=100);
```

```
kgamma(:,2)=k;
```

```
cgamma(:,2)=c;
```

```
rbig(:,2)=cgamma(2:101,2). $\hat{(-gam)}$ ./(bet*cgamma(3:102,2). $\hat{(-gam)}$ ));
```

```
%-----
```

```
% 5. Graphs and plots for other endogenous variables
```

```
%-----
```

```
%Let N be the periods to plot
```

```
N=40;
```

```
% The following equation compute the other endogenous variables use in the plots below
```

```
% Since they are function of capital and consumption, so we can compute them from the solved
```

```
% model above.
```

```
% These equations were taken from page 371 of RMT3
```

```
rbig0=1/bet;
```

```

nq0=alpha*A*k0^(alpha-1);
nq=alpha*A*kgamma(1:100,:).^(alpha-1);
wq0=A*k0^alpha-k0*alpha*A*k0^(alpha-1);
wq=A*kgamma(1:100,:).^alpha-kgamma(1:100,:).*alpha*A.*kgamma(1:100,:).^(alpha-1);

%Now we plot the responses of the endogenous variables to the shock.
x=0:N-1;
figure(1)

% subplot for capital 'k'
subplot(2,3,1)
plot(x,[k0*ones(N,1)],'-k', x,kgamma(1:N,1),'k',x,kgamma(1:N,2),'-k','LineWidth',1.5)
% note the timing: we lag capital to correct for syntax
title('k','FontSize',12)
set(gca,'FontSize',12)

% subplot for consumption 'c'
subplot(2,3,2)
plot(x,[c0*ones(N,1)],'-k', x,cgamma(2:N+1,1),'k',x,cgamma(2:N+1,2),'-k','LineWidth',1.5)
title('c','FontSize',12)
set(gca,'FontSize',12)

% subplot for cost of capital 'R_bar'
subplot(2,3,3)
plot(x,[rbig0*ones(N,1)],'-k', x,rbig(1:N,1),'k',x,rbig(1:N,2),'-k','LineWidth',1.5)
title('\overline{R}','interpreter','latex','FontSize',12)
set(gca,'FontSize',12)

% subplot for rental rate 'eta'
subplot(2,3,4)
plot(x,[nq0*ones(N,1)],'-k', x,nq(1:N,1),'k',x,nq(1:N,2),'-k','LineWidth',1.5)
title('\eta','FontSize',12)
set(gca,'FontSize',12)
legend('\gamma=0.2','\gamma=2')

% subplot for the experiment proposed

```

```

subplot(2,3,5)
plot([0:9],oo_.exo_simul(1:10,3),'k','LineWidth',1.5);
hold on;
plot([10:N-1],oo_.exo_simul(11:N,3),'k','LineWidth',1.5);
hold on;
plot(x,[g0*ones(N,1)],'-k','LineWidth',1.5)
title('g','FontSize',12)
axis([0 N -.1 .5])
set(gca,'FontSize',12)

```




Figure 16: Response to foreseen once-and-for-all increase in g at $t = 10$

5.3.6. Impact of an foreseen once-and-for-all increase in g on yield curves

Impact of an foreseen once-and-for-all increase in g at $t = 10$ on yield curves. For this section we use some additional formulas

$$q_t = \frac{\beta^t c_t^{-\gamma}}{c_0^{-\gamma}} \quad (20)$$

$$y_{i,t} = -\frac{1}{i} \log\left(\frac{q_{t+i}}{q_t}\right) \quad (21)$$

%
% 5. Graphs and plots for other endogenous variables

```

%-----
%Let N be the periods to plot
N=40; % The following equation compute the other endogenous variables use in the
plots below
% Since they are function of capital and consumption, so we can compute them from
the solved
% model above.

% These equations were taken from page 333 of RMT2
rbig0=1/bet;
rbig=c(2:101). $\hat{-gam}$ ./(bet*c(3:102). $\hat{-gam}$ ));
nq0=alpha*A*k0( $\hat{alpha-1}$ );
nq=alpha*A*k(1:100).( $\hat{alpha-1}$ );
wq0=A*k0 $\hat{alpha-k0*alpha}$ *A*k0( $\hat{alpha-1}$ );
wq=A*k(1:100). $\hat{alpha-k(1:100)}$ .*alpha*A.*k(1:100).( $\hat{alpha-1}$ );

% computing the price system : q and steady state price system qs(t)=beta $\hat{t}$ 
%q(2) is normalized to 1
%qs(1) is normalized to 1
for n=1:102
q(n)=bet $\hat{n}$ *c(n).( $\hat{-gam}$ )/(bet $\hat{2}$ *c(2)( $\hat{-gam}$ ));
qs(n)=bet $\hat{n}$ *c(1).( $\hat{-gam}$ )/(bet*c(1)( $\hat{-gam}$ ));
end

% computing the short term interest rates

rsmall=-log(q(3:102)./q(2:101));
rsmall0=rbig0-1;

% computing the term structure for n=1 to 40 for all periods

for t=1:62
for i=1:40
y(i,t)=log(q(i+t)/q(t))/-i;
end
end

```


%Now we plot the yield curves and other responses of the endogenous variables to the shock.

```
figure(1)
x=0:N-1;
% subplot for consumption 'c'
subplot(2,3,1)
plot(x,c0*ones(N,1),'-k', x,c(2:N+1),'k','LineWidth',1.5)
title('c','FontSize',12)
set(gca,'FontSize',12)

% subplot for consumption 'q'
subplot(2,3,2)
plot(x,qs(1:N),'-k', x,q(2:N+1),'k','LineWidth',1.5)
title('q','FontSize',12)
set(gca,'FontSize',12)

% subplot for rate of interest 'r'
subplot(2,3,3)
plot(x,[rsmall0*ones(N,1)],'-k', x,rsmall(1:N),'k','LineWidth',1.5)
title('r','FontSize',12)
set(gca,'FontSize',12)

% subplot for yield curves at t=0,t=10,t=60
subplot(2,3,4)
plot(x,y(:,2),'k', x,y(:,11),'-k',x,y(:,61),'-k','LineWidth',1.5)
title('','FontSize',12)
set(gca,'FontSize',12)
axis fill;
legend('t=0','t=10','t=60')

% subplot for the experiment proposed
subplot(2,3,5)
plot([0:9],oo_.exo_simul(1:10,3),'k','LineWidth',1.5);
hold on;
plot([10:N-1],oo_.exo_simul(11:N,3),'k','LineWidth',1.5);
hold on;
```

```

plot(x,[g0*ones(N,1)],'-k','LineWidth',1.5)
title('g','FontSize',12)
axis([0 N -.1 .5])
set(gca,'FontSize',12)

```




Figure 17: Response to foreseen once-and-for-all increase in g at $t = 10$. Yield curves for $t = 0$ (solid line), $t = 10$ (dashdotted line) and $t = 60$ (dashed line); term to maturity is on the x axis for the yield curve, time for the other panels

5.4. The model with elastic labor supply

In this section we modify the previous model by allowing for elastic labor supply. Now the representative agent maximizes

$$\sum_{t=0}^{\infty} \beta_t \log c_t + B(1 - n_t)$$

where B is substantially greater than 1 to assure an interior solution $n \in (0, 1)$ for labor supply.

Feasible allocations satisfy

$$g_t + c_t + k_{t+1} \leq A_t k_t^\alpha n_t^{1-\alpha} + (1 - \delta)k_t$$

The household budget constraint is

$$\sum_{t=0}^{\infty} \{q_t(1 + \tau_{c,t}) + q_t[k_{t+1} - (1 - \delta)k_t]\} \leq \sum_{t=0}^{\infty} q_t \{ \eta_t(1 - \tau_{k,t})k_t + w_t(1 - \tau_{n,t}) \}$$

The following conditions characterize the endogenous variables in the model equilibrium:

$$c_t = A_t k_t^\alpha n_t^{1-\alpha} + (1 - \delta)k_t - k_{t+1} - g_t \quad (22)$$

$$c_t^{-1} = \beta c_{t+1}^{-1} \left(\frac{1 + \tau_{c,t}}{1 + \tau_{c,t+1}} \right) [(1 - \delta) + (1 - \tau_{k,t+1}) A_t \alpha k_{t+1}^{\alpha-1} n_{t+1}^{1-\alpha}] \quad (23)$$

$$B c_t (1 + \tau_{c,t}) = (1 - \tau_{n,t}) [(1 - \alpha) A_t k_t^\alpha n_t^{-\alpha}] \quad (24)$$

5.4.1. Elastic Labor supply : Unforeseen once-and-for-all increase in g at t=0

An unforeseen once-and-for-all increase in g to .2 at t = 0 (Figure 18).

```
%
%-----
% Experiment : Unforeseen increase in g at t=0
%-----

% This program replicates figure 11.9.1 from chapter 11 of RMT3 by Ljungqvist and
Sargent

% Dynare records the endogenous variables with the following convention. Say
N is the number of simulations(sample)
%Index 1 : Initial values (steady state)
%Index 2 to N+1 : N simulated values
%Index N+2 : Temnial Value (Steady State)
% Warning: we align c, k, and the taxes to exploit the dynare syntax. In Dynare the
timing of the variable reflects the date
%when the variable is decided. For instance the capital stock for time 't' is decided
in time 't-1'(end of period). So a statement like
% k(t+1) = i(t) + (1-del)*k(t) would translate to " k(t) = i(t) +(1-del)*k(t-1)" in
the code.

%-----
```

```
% 1. Defining variables
```

```
%-----
```

```
%Declares the endogenous variables consumption ('c') capital stock (k);
```

```
var c k n;
```

```
%declares the exogenous variables consumption tax ('tauc'), capital tax('tauk'),  
government spending('g')
```

```
varexo tauc tauk taun g;
```

```
parameters bet gam del alpha A B;
```

```
%-----
```

```
% 2. Calibration and alignment convention
```

```
%-----
```

```
bet=.95; % discount factor
```

```
gam=1; % CRRA parameter
```

```
del=.2; % depreciation rate
```

```
alpha=.33; % capital's share
```

```
A=1; % productivity
```

```
B=3; % coefficient on leisure
```

```
% Alignment convention:
```

```
% g tauc tauk taun are now columns of ex_. Because of a bad design decision
```

```
% the date of ex_(1,:) doesn't necessarily match the date in endogenous variables.
```

```
Whether they match depends
```

```
% on the number of lag periods in endogenous versus exogenous variables.
```

```
% These decisions and the timing conventions mean that
```

```
% k(1) records the initial steady state, while k(102) records the terminal steady state  
values.
```

```
% For  $j > 1$ , k(j) records the variables for j-1th simulation where the capital stock  
decision
```

```
%taken in j-1 th simulation i.e stock at the beginning of period j.
```

```
% The variable ex_ also follows a different timing convention i.e ex_(j,:) records the  
value of exogenous variables in the j th simulation.
```

```
%The jump in the government policy is reflected in ex_(10,1) for instance.
```

```

%-----
% 3. Model
%-----

model;

%Feasibility
k=A*k(-1)^alpha*n^(1-alpha)+(1-del)*k(-1)-c-g;

%Euler equation
c^(1-gam)=bet*(c(+1)^(1-gam))*((1+tauc)/(1+tauc(+1)))*((1-del)+(1-tauk(+1))^alpha*A*k^(alpha-1)*n(+1)^(1-alpha));

%Consumption leisure choice
B/c^(1-gam)=(1-taun)*((1-alpha)*A*k(-1)^(alpha)*n^(1-alpha))*(1+tauc)^1;

end;

%-----
% 4. Computation
%-----

initval;
k=1.5;
c=0.6;
g = .2;
n=1.02;
tauc = 0;
tauk = 0;
taun=0;
end;
steady; % put this in if you want to start from the initial steady state, comment it
out to start from the indicated values

endval; % The following values determine the new steady state after the shocks.
k=1.5;

```

```

c=0.6;
g =.4;
n=1.02;
tauc =0;
tauk =0;
taun=0;
end;

steady; % We use steady again and the enval provided are initial guesses for dynare
to compute the ss.

% now solve the model
simul(periods=100);

% Compute the initial steady state for consumption to later do the plots.
c0=c(1);
k0 = k(1);
n0=n(1);
% g is in ex_(:,1) since it is stored in alphabetical order
g0 = .2;

%-----
% 5. Graphs and plots for other endogenous variables
%-----

%Let N be the periods to plot
N=40;

% The following equation compute the other endogenous variables use in the plots
below
% Since they are function of capital and consumption, so we can compute them from
the solved
% model above.
% These equations were taken from page 371 of RMT3
rbig0=1/bet;
rbig=c(2:101).1-gam./(bet*c(3:102).1-gam);
nq0=alpha*A*k0alpha-1*n01-alpha;

```

```

nq=alpha*A*k(1:100). $\hat{\alpha}$ *(n(2:101). $\hat{1-\alpha}$ );
wq0=A*(1-alpha)*k0 $\hat{\alpha}$ *n0 $\hat{-\alpha}$ ;
wq=(1-alpha)*A*k(1:100). $\hat{\alpha}$ *(n(2:101). $\hat{-\alpha}$ );

%Now we plot the responses of the endogenous variables to the shock.
x=0:N-1;
figure(1)

% subplot for capital 'k'
subplot(2,3,1)
plot(x,[k0*ones(N,1)],'-k', x,k(1:N),'k','LineWidth',1.5) % note the timing: we lag
capital to correct for syntax
title('k','FontSize',12)
set(gca,'FontSize',12)

% subplot for consumption 'c'
subplot(2,3,2)
plot(x,[c0*ones(N,1)],'-k', x,c(2:N+1),'k','LineWidth',1.5)
title('c','FontSize',12)
set(gca,'FontSize',12)

% subplot for consumption 'n'
subplot(2,3,3)
plot(x,[n0*ones(N,1)],'-k', x,n(2:N+1),'k','LineWidth',1.5)
title('n','FontSize',12)
set(gca,'FontSize',12)

% subplot for cost of capital 'R_bar'
subplot(2,3,4)
plot(x,[rbig0*ones(N,1)],'-k', x,rbig(1:N),'k','LineWidth',1.5)
title('\overline{R}','interpreter','latex','FontSize',12)
set(gca,'FontSize',12)

% subplot for wage rate 'w'
subplot(2,3,5)
plot(x,[wq0*ones(N,1)],'-k', x,wq(1:N),'k','LineWidth',1.5)
title('w','FontSize',12)

```

```
set(gca,'FontSize',12)
```

```
% subplot for the experiment proposed
```

```
subplot(2,3,6)
```

```
plot(x,[g0*ones(N,1)],'-k', x,oo_.exo_simul(1:N,4),'k','LineWidth',1.5)
```

```
title('g','FontSize',12)
```

```
axis([0 40 -.1 .5])
```

```
set(gca,'FontSize',12)
```




Figure 18: Elastic labor supply: response to unforeseen increase in g at $t = 0$

5.4.2. Elastic Labor supply : Foreseen once-and-for-all increase in τ_n at $t=0$

A foreseen once-and-for-all increase in τ_n to 20 percent at $t = 0$ (Figure 19).

```
initval;
```

```
k=1.5;
```

```
c=0.6;
```

```
g = .2;
```

```
n=1.02;
```

```
tauc = 0;
```

```
tauk = 0;
```



```

taun=0;
end;
steady;
endval;
k=1.5;
c=0.6;
g=.2;
n=1.02;
tauc=0;
tauk=0;
taun=0.2;
end;
steady;

```




Figure 19: Elastic labor supply: response to unforeseen increase in τ_n at $t = 0$

5.4.3. Elastic Labor supply : Foreseen once-and-for-all increase in τ_n at $t=10$

A foreseen once-and-for-all increase in τ_n to 20 percent at $t=10$ (Figure 20).

```

initval;
k=1.5;

```

```

c=0.6;
g = .2;
n=1.02;
tauc = 0;
tauk = 0;
taun=0;
end;
steady;
endval;
k=1.5;
c=0.6;
g =.2;
n=1.02;
tauc =0;
tauk =0;
taun=0.2;
end;
steady;
shocks;
var taun;
periods 1:10;
values 0;
end;

```




Figure 20: Elastic labor supply: response to unforeseen increase in τ_n at $t = 10$

6. The demand for money during hyperinflations

Sargent (1977) specified and estimated a joint stochastic process for money growth and inflation that was designed to make the adaptive expectations model of Cagan (1956) consistent with rational expectations. The model has the form

$$m_t - p_t = \alpha \pi_t + u_t \quad (25)$$

$$\pi_t = \frac{1 - \lambda}{1 - \lambda L} (p_t - p_{t-1}) = \frac{1 - \lambda}{1 - \lambda L} x_t \quad (26)$$

$$\mu_t = m_t - m_{t-1} \quad (27)$$

with $\alpha < 0$, m being the log of money supply, π the expectation of inflation, p the log of the price level and u an error term whose specification is described below. The hypothesis of rational expectations imposes

$$\pi_t = E_t x_{t+1} \quad (28)$$

Sargent (1977) reverse engineered a process for x_t such that Cagan's adaptive expectations scheme is equal to rational expectations. The following conditions ensure the rationality of Cagan's expectations scheme

$$u_t = u_{t-1} + \eta_t \quad (29)$$

$$\mu_t = \frac{1 - \lambda}{1 - \lambda L} x_t + \epsilon_t \quad (30)$$

η and ϵ are both serially uncorrelated random error terms with mean zero and variances σ_η^2 and σ_ϵ^2 , respectively. For estimating the model, it is convenient to represent it in the following form

$$\begin{pmatrix} x_t \\ \mu_t \end{pmatrix} = \begin{pmatrix} 1 & 0 \\ 1 - \lambda & \lambda \end{pmatrix} \begin{pmatrix} x_{t-1} \\ \mu_{t-1} \end{pmatrix} + \begin{pmatrix} a_{1,t} \\ a_{2,t} \end{pmatrix} - \lambda I \begin{pmatrix} a_{1,t-1} \\ a_{2,t-1} \end{pmatrix} \quad (31)$$

The a terms are linked to the random terms above via

$$\begin{pmatrix} a_{1,t} \\ a_{2,t} \end{pmatrix} = \begin{pmatrix} 1 & -1 \\ 1 + a(1 - \lambda) & -(1 - \lambda) \end{pmatrix} (\lambda + (1 - \lambda)\alpha)^{-1} \begin{pmatrix} \epsilon_t \\ \eta_t \end{pmatrix} \quad (32)$$

The code below estimates representation (31) by maximum likelihood:

```
% this program estimates the model in
% "The Demand for Money during Hyperinflations under Rational Expectations: I"
% by T. Sargent, IER 1977 using maximum likelihood
% variables are defined as follows:
% x=p_t-p_t-1, p being the log of the price level
% mu=m_t-m_t-1, m being the log of money supply
% note that in contrast to the paper eta and epsilon have variance 1 (they are
% multiplied by the standard deviations)

var x mu a1 a2;

varexo epsilon eta;

parameters alpha lambda sig_eta sig_epsilon;

lambda=.5921;
alpha=-2.344;
sig_eta=.001;
sig_epsilon=.001;

model;
x=x(-1)-lambda*a1(-1)+(1/(lambda+alpha*(1-lambda)))*sig_epsilon*epsilon-(1/(lambda+alpha*(1-
lambda)))*sig_eta*eta;
mu=(1-lambda)*x(-1)+lambda*mu(-1)-lambda*a2(-1)+(1+alpha*(1-lambda))/(lambda+alpha*(1-
```

```

lambda))*sig_epsilon*epsilon-(1-lambda)/(lambda+alpha*(1-lambda))*sig_eta*eta;
a1=(1/(lambda+alpha*(1-lambda)))*sig_epsilon*epsilon-(1/(lambda+alpha*(1-lambda)))*sig_eta*eta;
a2=(1+alpha*(1-lambda))/(lambda+alpha*(1-lambda))*sig_epsilon*epsilon-(1-lambda)/(lambda+alpha*(1-lambda))*sig_eta*eta;
end;

steady;

shocks;

var eta;
stderr 1;
var epsilon;
stderr 1;
end;

estimated_params;
% ML estimation setup
% parameter name, initial value, boundaries_low, ..._up;
lambda, .5, 0.25, 0.75;
alpha, -2, -8, -0.1;
sig_eta, .0001, 0.0001, 0.3;
sig_epsilon, .0001, 0.0001, 0.3;
end;

varobs mu x;

estimation(datafile=cagan_data,first_obs=1,nobs=34,mh_replic=0,mode_compute=4,mode_check=0);

```

The data for the estimation are from Cagan (1956) for Germany¹. The shape of the log-likelihood function is plotted in figure 21.

¹The data are demeaned, allowing us to ignore constant terms in the model




Figure 21: the Log-Likelihood function for the hyperinflation model

h

Table 8: Results from Maximum likelihood

Parameter	Estimate	s.d.	t-stat.
λ	0.6601	0.0542	12.1709
α	-4.6436	2.8468	1.6312
σ_{η}	0.2355	0.1992	1.1824
σ_{ϵ}	0.0728	0.0101	7.2353

Next we estimate this model using Bayesian techniques. For this we replace the last part of the code above with

```
estimated_params;
% Bayesian setup
lambda, uniform_pdf, 0.68, .5;
alpha, uniform_pdf, -5, 2;
sig_eta, uniform_pdf, .5, 0.25;
sig_epsilon, uniform_pdf, .5, 0.25;
end;

varobs mu x;

estimation(datafile=cagan_data,nobs=34,mh_replic=25000,mh_nblocks=1,mh_jscale=1,mode_com
```

Note that the priors we have chosen are reasonably loose, but given that there are only 34 observations we have opted to use somewhat informative priors. The posteriors and priors are plotted in figure 22.




Figure 22: Posteriors and priors for the hyperinflation model

7. Solving and estimating Hall's permanent income model

This note describes a suite of Dynare programs to simulate and estimate the permanent income model of Hall(1988).

7.1. The model

A consumer faces the following decision problem:

$$\underbrace{\max}_{c_t, k_t} -\frac{1}{2} \sum_{t=0}^{\infty} \beta^t (c_t - b_t)^2$$

subject to the following constraints

$$c_t + k_t = R_t k_{t-1} + d_t$$

$$d_t = \rho d_{t-1} + (1 - \rho) \mu_d + \epsilon_d$$

$$b_t = \rho_b b_{t-1} + (1 - \rho_b) \mu_b + \epsilon_b$$

We assume that $\beta R = 1$ and that there is no depreciation, which leads to the following law of motion for investment:

$$i_t = k_t - k_{t-1}$$

From the Lagrangian for this problem one obtains the following first order conditions:

$$\mu_{c,t} = \mu_{c,t+1}$$

$$\mu_{c,t} = b_t + c_t$$

where $\mu_{c,t}$ is the marginal utility of consumption or the Lagrange multiplier.

Note that this problem has a unit root and so there exist no steady state. However, we can supply ONE solution corresponding to an initial k that we give to Dynare.

The following Dynare code simulates data from this model:

```
var c k mu_c b d in;  
varexo e_d e_b;
```

```
parameters R rho rho_b mu_b mu_d;
```



```

R=1.05;
% rho=0.9;
rho = 0;
mu_b=30;
mu_d=5;
rho_b = 0;

model(linear);

c+k = R*k(-1) + d;
mu_c = b - c;
mu_c=mu_c(+1);
d= rho*d(-1)+ mu_d*(1-rho) + e_d;
b=(1-rho_b)*mu_b+rho_b*b(-1)+e_b;
in = k - k(-1);
end;

%With a unit root, there exists no steady state. Use the following trick.
%Supply ONE solution corresponding to the initial k that you named.

initval;
d=mu_d;
k=100;
c = (R-1)*k +d;
mu_c=mu_b-c;
b=mu_b;
end;

shocks;
var e_d;
stderr 1;
var e_b;
stderr 1;
end;

steady;
check;

```

```

stoch_simul(order=1, periods=500, irf=10);
save data_hall c in;

```




Figure 23: A shock to ϵ_b

Next, we proceed to estimate a subset of the parameters of the model by ML using simulated data obtained from the previous code. To do so we add the following few lines of code. Notice how we use the command `unit root vars` to initialize the Kalman filter with a diffuse prior. We do this because k is not stationary – Hall’s model with $\beta R = 1$ has a solution for k that has a unit root and that is co-integrated with consumption.

```

estimated_params;
% ML estimation setup
% parameter name, initial value, boundaries_low, ..._up;
rho, 0.9, 0, 1;
R, 1.05, 0, 1.5;
end;
varobs c in;

```

```

estimation(datafile=data_hall,first_obs=101,nobs=200,mh_replic=0,mode_compute=4,mode_check

```




Figure 24: priors and posteriors from the ML estimation of Hall's model

We proceed to estimate the same subset of the parameters using Bayesian techniques: This is implemented in Dynare by adding the following few lines of code.

```
estimated_params;
rho,beta_pdf, 0.1, 0.2;
R,norm_pdf, 1.02, 0.05;
end;
varobs c in;
```

```
estimation(datafile=data_hall,first_obs=101,nobs=200,mode_compute=6,mh_replic=1000,mh_nblo
```

8. Solving and estimating Ryoo and Rosen's (2004) model of the market for engineers

This section describes a suite of Dynare programs to simulate and estimate a model of Aloysius Siow (1984) and Jaewoo Ryoo and Sherwin Rosen (2004) of the labor market for engineers.

8.1. The model

Ryoo and Rosen studied a partial equilibrium model determining a stock of 'engineers' N_t ; the number of new entrants into engineering school, s_t ; and the wage level w_t of engineers.² It takes k periods of schooling to become an engineer. The model consists of the following components:

The present value of wages of an engineer:

$$P_t = (1 - \delta)\beta P_{t+1} + (1 - \delta)^k \beta^k w_{t+k}$$

A supply of new engineering students:

$$s_t = a_0 + a_1 P_t + e_{s,t}$$

A time-to-build structure of the education process:

$$N_t = (1 - \delta)N_{t-1} + s_{t-k}$$

A demand curve for engineers:

$$N_t = d_0 - d_1 w_t + e_{d,t}$$

The following Dynare code simulates data from this model:

```
% Rosen schooling model
% The model is the one Sherwin Rosen showed Sargent in Sargent's Chicago office.
% The equations are

% s_t = a0 + a1*P_t + e_st ; flow supply of new engineers
% N_t = (1-delta)*N_t-1 + s_t-k ; time to school engineers
% N_t = d0 - d1*W_t +e_dt ; demand for engineers
% P_t = (1-delta)*bet P_(t+1) + beta^k*W_(t+k); present value of wages of an
engineer

var s N P W;
varexo e_s e_d;

parameters a0 a1 delta d0 d1 bet k;
a0=10;
```

²Sherwin Rosen showed this version of the model to Sargent in Chicago in fall 1995.

```

a1=1;
d0=1000;
d1=1;
bet=.99;
delta=.02;

model(linear);
s=a0+a1*P+e_s; % flow supply of new entrants
N=(1-delta)*N(-1) + s(-4); % evolution of the stock
N=d0-d1*W+e_d; % stock demand equation
P=bet*(1-delta)*P(+1) + bet*delta*(1-delta)*delta*W(+4); % present value of wages
end;

initval;
s=0;
N=0;
P=0;
W=0;
end;

shocks;
var e_d;
stderr 1;
var e_s;
stderr 1;
end;

steady;
check;

stoch_simul(order=1, periods=500, irf=10);
save data_rosen s N P W;

```

Figure 25 plots the response of endogenous variables to ϵ_s .




Figure 25: A shock to $\epsilon_{s,t}$

Note that in this code we set $k = 4$. Dynare does not let you use a parameter as lag length.

Next, we estimate a subset of the parameters of the model by ML using simulated data obtained from the previous code. To do so we add the following lines of code.

```
estimated_params;
a1, .5, -10, 10;
d1, .5, -20, 40;
stderr e_d, .5, .05, 5;
stderr e_s, .3, .05, 6;
end;
% these are the ranges for the parameters
varobs W N;
estimation(datafile=data_rosen,first_obs=101,nobs=200,mh_replic=0,mode_compute=4,mode_check=0)
```

Figure 26 plots the log-likelihood function for the parameters.




Figure 26: the Log-Likelihood for Ryoo and Rosen's model

We proceed to estimate the same subset of the parameters using Bayesian techniques. We implemented this in Dynare by adding the following lines of code.

```

estimated_params;
a1, gamma_pdf, .5, .5;
d1, gamma_pdf, 2, .5;
stderr e_d, inv_gamma_pdf, .5, 30;
stderr e_s, inv_gamma_pdf, .2, 30;
% The syntax for to input the priors is the following:
% variable name, prior distribution, parameters of distribution.
end;
varobs W N;
estimation(datafile=data_rosen, first_obs=101, nobs=200, mh_replic=10000, mh_nblocks=2,
mh_jscale=2, mode_compute=0, mode_file=rosenestimateML_mode);

```




Figure 27: the Log-Likelihood for Ryoo and Rosen's model

h

Table 9: Results from Bayesian Estimation

Parameter	Prior mean	Posterior mean	90% HPD	interval	Prior	pst.dev.
a_1	0.500	0.9753	0.7429	1.2383	gamm	0.5000
d_1	2.000	1.0638	0.8841	1.2800	gamm	0.5000
ϵ_d	0.500	0.9945	0.7655	1.1704	invg	30.0000
ϵ_s	0.200	0.9306	0.8120	1.0245	invg	30.0000

9. Estimation of Bansal and Yaron(2004) consumption growth process

Let consumption growth have the state space representation

$$x_t = \rho x_{t-1} + \sigma_x \epsilon_{x,t}$$

$$\Delta c_t = \mu_c + x_{t-1} + \sigma_c \epsilon_{c,t}$$

where the shocks $\epsilon_{c,t}$ and $\epsilon_{x,t}$ are i.i.d. standard normal. We use quarterly growth rate of per capita U.S. consumption over the period 1947.2-2003.3 to estimate the four unknown parameters. The following Dynare code does this by maximum likelihood.

```
var x y;
varexo e_x e_u;

parameters rho sig_x sig_u mu_y;

rho = .98;
mu_y=.015;
sig_x=0.00025;
sig_u=.0078;

model(linear);
x=rho*x(-1) + sig_x*e_x;
y=mu_y + x(-1) + sig_u*e_u;
end;

initval;
x=0;
y=mu_y;
end;

steady;

shocks;
var e_x;
stderr 1;
```

```

var e_u;
stderr 1;
end;

estimated_params;
% ML estimation setup
% parameter name, initial value, boundaries_low, ..._up;
rho, 0, -0.99, 0.999; % use this for unconstrained max likelihood
% rho, .98, .975, .999 ; % use this for long run risk model
% sig_x, .0004,.0001,.05 ; % use this for the long run risk model
sig_x, .0005, .000000000001, .01; % use this for unconstrained max likelihood
sig_u, .007,.001, .1;
mu_y, .014, .0001, .04;

end;

varobs y;
estimation(datafile=data_consRicardoypg,first_obs=1,nobs=227,mh_replic=0,mode_compute=4,mo

```

Table 10 reports the estimated parameter from Dynare.

h

Table 10: Results from Maximum likelihood

Parameter	Estimate	s.d.	t-stat.
ρ	0.5623	0.1926	2.9195
μ_y	0.0056	0.0005	10.7599
σ_x	0.0030	0.0011	2.6935
σ_u	0.0040	0.0007	5.4546

Figure 29 also plots the log-likelihood function along several dimensions.




Figure 28: the Log-Likelihood function for the consumption growth model

Next, we estimate the same model using Bayesian techniques and two different sets of priors. The first set of priors are tight around the calibrated values used in Bansal and Yaron (2004). To accomplish this, we can simply change the following lines of code from our MLE program above.

```

estimated_params;
rho, beta_pdf, .98, .01;
mu_y, uniform_pdf, .005, .0025;
sig_u, inv_gamma_pdf, .003, inf;
sig_x, inv_gamma_pdf, .003, inf;
% The syntax for to input the priors is the following:
% variable name, prior distribution, parameters of distribution.
end;

```

Table 11: Results from Bayesian Estimation


Parameter	Prior mean	Posterior mean	90% HPD interval	Prior	pst.dev.
ρ	0.980	0.9711	0.9513 0.9905	beta	0.0100
μ_y	0.005	0.0051	0.0019 0.0088	unif	0.0025
σ_u	0.003	0.0048	0.0043 0.0053	invg	Inf
σ_x	0.003	0.0012	0.0007 0.0016	invg	Inf

varobs y;

estimation(datafile=data_consRicardoypg,first_obs=1,nobs=227,mh_replic=5000,mh_nblocks=1,mh

Table 11 reports the estimated parameter from Dynare.

Figure 29 plots the prior and the posterior densities for the estimated parameters. The posteriors are plotted in black and priors in gray. A vertical green line denotes a posterior mean.

**Figure 29:** Prior and Posteriors for the Bansal and Yaron (2004) endowment process

Next, we set vague priors for ρ and for μ_y .

rho, beta_pdf, .5, .2;

mu_y, uniform_pdf, .005, .0025;

sig_u, inv_gamma_pdf, .003, inf;

Table 12: Results from Bayesian Estimation

Parameter	Prior mean	Posterior mean	90% HPD interval	Prior	pst.dev.
ρ	0.500	0.5044	0.2604 0.7198	beta	0.0200
μ_y	0.004	0.0056	0.0046 0.0064	unif	0.003
σ_u	0.003	0.0035	0.0015 0.0048	invg	Inf
σ_x	0.003	0.0034	0.0020 0.0053	invg	Inf

`sig_x, inv_gamma_pdf, .003, inf;`

Table 12 reports the estimated parameter from Dynare.

Figure 30 plots the prior and the posterior densities for the estimated parameters. Notice that with a more diffuse prior on ρ , the posteriors for μ_y are tighter.




Figure 30: Prior and Posteriors for the Bansal and Yaron (2004) endowment process

10. Robust permanent income and pricing

Hansen, Sargent, and Tallarini (1999) study consumption, saving, and security market prices in a permanent income model in which the agents mistrust their specification of the economic environment. The preferences of a representative consumer are defined in terms of the following risk-sensitivity recursion:

$$U_t = u_t + \beta R_t(U_{t+1}) \quad (33)$$

$$R_t(U_{t+1}) = \frac{2}{\sigma} \log[\exp(\frac{\sigma U_{t+1}}{2})] \quad (34)$$

with u_t being the per-period utility function. Note that $\sigma = 0$ corresponds to Von Neumann-Morgenstern preferences. The authors use a linear quadratic framework described below. Furthermore, they show that for a given (β, σ) pair there exists another pair $(\hat{\beta}, 0)$ that gives the same equilibrium quantities. This observation simplifies solution and estimation of the model significantly. For the $\sigma = 0$ case and HST's linear-quadratic habit persistence model, preferences become

$$U_0 = E_0 \sum_{t=0}^{\infty} \beta^t (-(s_t - b)^2) \quad (35)$$

with b being a subsistence point and s defined as

$$s_t = (1 + \lambda)c_t - \lambda h_{t-1} \quad (36)$$

$$h_t = \delta_h h_{t-1} + (1 - \delta_h)c_t \quad (37)$$

where c_t is current period consumption and h_{t-1} the habit stock. There is a linear production technology:

$$c_t + i_t = \gamma k_{t-1} + d_t \quad (38)$$

$$k_t = \delta_k k_{t-1} + i_t \quad (39)$$

$$d_t = \mu_d + \hat{d}_t + \bar{d}_t \quad (40)$$

$$\bar{d}_t = (\phi_1 + \phi_2)\bar{d}_{t-1} - \phi_1\phi_2\bar{d}_{t-2} + c_{\bar{d}}e_{1,t} \quad (41)$$

$$\hat{d}_t = (\alpha_1 + \alpha_2)\hat{d}_{t-1} - \alpha_1\alpha_2\hat{d}_{t-2} + c_{\hat{d}}e_{2,t} \quad (42)$$

$e_{1,t}$ and $e_{2,t}$ are standard normal iid shocks. Finally, we define the gross return on capital as

$$R \equiv \delta_k + \gamma \quad (43)$$

and impose

$$R\beta = 1 \quad (44)$$

To solve the planning problem, we formulate a Lagrangian with multipliers $2\beta^t\mu_{s,t}$ on (36), $2\beta^t\mu_{h,t}$ on (37) and $2\beta^t\mu_{i,t}$ on (38) (after substituting for investment i_t).

The first order necessary conditions are:

$$\mu_{s,t} = b - s_t \quad (45)$$

$$\mu_{c,t} = (1 + \lambda)\mu_{s,t} + (1 - \delta_h)\mu_{h,t} \quad (46)$$

$$\mu_{h,t} = \beta E_t[\delta_h \mu_{h,t+1} - \lambda \mu_{s,t+1}] \quad (47)$$

$$\mu_{c,t} = \mu_{c,t+1} \quad (48)$$

plus (36)-(39). The following code estimates the model using maximum likelihood:

```
% Estimates the Hansen Sargent and Tallarini model by maximum likelihood.

var s c h k i d dhat dbar mus muc muh gamma R;
varexo e_dhat e_dbar;

parameters lambda deltah deltak mud b bet phi1 phi2 cdbar alpha1 alpha2 cd-
hat;
bet=0.9971;
deltah=0.682;
lambda=2.443;
alpha1=0.813;
alpha2=0.189;
phi1=0.998;
phi2=0.704;
mud=13.710;
cdhat=0.155;
cdbar=0.108;
b=32;
deltak=0.975;

model(linear);
R=deltak+gamma;
R*bet=1;
s=(1+lambda)*c-lambda*h(-1);
h=deltah*h(-1)+(1-deltah)*c;
k=deltak*k(-1)+i;
c+i=(1/bet-deltak)*k(-1)+d;
mus=b-s;
```

```

muc=(1+lambda)*mus+(1-deltah)*muh;
muh=bet*(deltah*muh(+1)-lambda*mus(+1));
muc=muc(+1);
d=mud+dbar+dhat;
dbar=(phi1+phi2)*dbar(-1) - phi1*phi2*dbar(-2) + cdbar*e_dbar;
dhat=(alpha1+alpha2)*dhat(-1) - alpha1*alpha2*dhat(-2) + cdhat*e_dhat;
end;

initval;
R=1/bet;
gamma=R-deltak;
d=mud;
k=100;
i=(1-deltak)*k;
c=gamma*k+d-i;
h=c;
s=c;
mus=b-s;
muh=lambda/(deltah-1/bet)*mus;
muc=(1+lambda)*mus+(1-deltah)*muh;
end;

steady(nocheck);

shocks;
var e_dhat;
stderr 1;
var e_dbar;
stderr 1;
end;

% stoch_simul(irf=0, periods=500);
% save dataHST c i;

estimated_params;
bet, .9971, 0, .99999;
deltah, 0.682, 0.1, 0.8;

```



```

lambda, 2.443, 0.1, 50;
alpha1, 0.813, 0.6, 0.99999;
alpha2, 0.189, 0.01, 0.5;
phi1, 0.998, 0, 0.99999;
phi2, 0.704, 0, 0.9;
mud, 13.710, 1, 50;
cdhat, 0.155, 0.05, 0.2;
cdbar, 0.108, 0.05, 0.2;
end;

varobs c i;
estimation(datafile=dataHST,first_obs=1,nobs=500,mode_compute=4,mode_check,diffuse_filter);

```

To estimate the model using Bayesian techniques simply replace the last part of the code with

```

estimated_params;
bet,uniform_pdf, .9499999999, 0.0288675134306;
deltah,uniform_pdf, 0.45, 0.202072594216; lambda,uniform_pdf,25.05,14.4048892163;
alpha1,uniform_pdf, 0.8, 0.115470053809; alpha2,uniform_pdf, 0.25, 0.144337567297;
phi1,uniform_pdf,0.8,0.115470053809; phi2,uniform_pdf, 0.5, 0.288675134595; mud,uniform_pdf,
24.5, 14.1450815951; cdhat,uniform_pdf,0.175,0.0721687836487; cdbar,uniform_pdf,
0.175, 0.0721687836487;
end;

varobs c i;
estimation(datafile=dataHST,nobs=500,mode_check,mh_replic=10000,mh_nblocks=1,mh_jscale=0.

```

The maximization routine used here (mode compute=4) is Chris Sims' csminwel. Note that this procedure uses a random number generator, so different runs of this code can lead to different estimates (the routine might get stuck in a local extremum). Thus, it is important to run the code several times.

11. Asset pricing with risks for the long run

11.1. Introduction

Following Bansal and Yaron (2004), we specify dividend and consumption growth as containing a small but highly persistent predictable component. These dynamics, when combined with Epstein and Zin (1989) preferences, help explain financial phenomena including the equity premium puzzle. What allows this specification to be successful is the ability of Epstein and Zin preferences to separate risk aversion from intertemporal elasticity of substitution, possibly allowing both to be larger than one. Dynare enables us to compute policy functions for the response of price-dividend ratios to an increase in the long-run predictable component of dividend and consumption growth. Using the specification just described, this impact turns out to be positive. In economic terms, the intertemporal substitution effect dominates the wealth effect. In response to higher expected growth rates, agents buy more assets and this increases the wealth to consumption ratio. In the standard power utility (constant relative risk aversion) model, the wealth effect dominates the substitution effect. Dynare shows that the policy functions for the latter specification display the opposite sign for the feedback of the predictable component of growth on price-dividend ratios. Hence the former specification helps explain high equity premia, because when the IES is larger than one, agents fear that a reduction in economic growth prospects will lower asset prices.

11.2. Setup of the model

Specify preferences as follows:

$$U_t = \left[(1 - \delta) C_t^{\frac{1-\gamma}{\theta}} + \delta \left(E_t [U_{t+1}^{1-\gamma}] \right)^{\frac{1}{\theta}} \right]^{\frac{\theta}{1-\gamma}} \quad (49)$$

where

$$\theta = \frac{1 - \gamma}{1 - \frac{1}{\Phi}} \quad (50)$$

Here γ reflects the degree of risk aversion, while Φ take into account intertemporal elasticity of substitution. When $\theta = 1$, preferences collapse to the standard CRRA case. Hence this model can be regarded as a general one.

There are N assets in this economy, whose one period ahead return is denoted as $R_{j,t+1}, \forall j = 1, \dots, N$. The share of investment in asset j is denoted as $\omega_{j,t}$. Hence there are $N - 1$ linearly independent elements in $\omega_t = [\omega_{1,t}, \omega_{2,t}, \dots, \omega_{N,t}]'$, since

portfolio shares must satisfy:

$$\sum_{j=1}^N \omega_{j,t} = 1, \forall t \quad (51)$$

The representative agent is endowed with an initial stock of the consumption asset A_0 that can be either consumed or invested in the N assets in the competitive market. Therefore, her wealth evolves according to:

$$A_{t+1} = (A_t - C_t)\omega'_t R_t \quad (52)$$

Following Epstein and Zin (1991), first order conditions to price an asset that entitles to a claim on a stream of consumption good imply:

$$v_{c,t} = E_t[\delta^\theta e^{-\frac{\theta}{\Phi}\Delta c_{t+1} + (\theta-1)r_{c,t+1}}(1 + v_{c,t+1})e^{\Delta c_{t+1}}] \quad (53)$$

where

$$\begin{aligned} \Delta c_t &= \log(C_t) - \log(C_{t-1}) \\ r_{c,t+1} &= \log R_{c,t+1} \\ R_{c,t+1} &= \frac{(1 + v_{c,t+1})e^{\Delta c_{t+1}}}{v_{c,t}} \\ v_{c,t} &= \frac{P_{c,t}}{C_t} \end{aligned}$$

Similarly, first order condition to price an asset that entitles to a dividend D_t in each period imply:

$$v_{d,t} = E_t[\delta^\theta e^{-\frac{\theta}{\Phi}\Delta d_{t+1} + (\theta-1)r_{d,t+1}}(1 + v_{d,t+1})e^{\Delta d_{t+1}}] \quad (54)$$

where

$$\begin{aligned} \Delta d_t &= \log(D_t - D_{t-1}) \\ v_{d,t} &= \frac{P_{d,t}}{D_t} \end{aligned}$$

We close the model, by specifying exogenous processes for consumption and dividend growth:

$$\Delta c_t = \mu_c + x_{t-1} + \epsilon_t \quad (55)$$

$$\Delta d_t = \mu_d + \lambda_{d,x}x_{t-1} + \epsilon_{d,t} \quad (56)$$

$$x_t = \rho_x x_{t-1} + \epsilon_{x,t} \quad (57)$$

where the shocks $\epsilon_{c,t}, \epsilon_{d,t}$ and $\epsilon_{s,t}$ are i.i.d. normal with mean zero and standard deviations σ_c, σ_d and σ_x , respectively. For stationarity $|\rho_x| < 1$. The economy is completely described by equations (53)-(57).

11.3. Calibration and simulation

We calibrate the parameters of the model in two different ways. Our guidelines in doing this are:

- use a reasonable risk aversion
- model the process of x as highly persistent

One important feature that, following the lines of Bansal and Yaron (2004), we would like to display is the different way in which x_t affects price-dividend ratios in the presence of Epstein and Zin utility functions. As we stressed before the channel through which a highly persistent and predictable component of consumption and dividend growth combined with Epstein-Zin preferences helps explaining the equity premium puzzle is the possibility of having intertemporal elasticity of substitution and risk aversion greater than one at the same time. Hence we want to experiment both a calibration where this condition is met and a standard constant relative risk aversion preferences environment, where risk aversion greater than one implies IES smaller than unity. The latter specification is a particular case in our model, as it only requires to put $\theta = 1$.

11.3.1. CRRA specification

We calibrate parameters as in Table 13. The numbers are the same as in Bansal and Yaron (2004) except for risk IES, that is now the inverse of risk aversion.

Table 13: parameter calibration. The value of θ is calibrated to satisfy equation (50).

Parameter	Calibration
Φ	0.133
θ	1
γ	7.5
δ	0.99
μ_c	0.0015
σ_c	0.0078
μ_d	0.0015
$\lambda_{d,x}$	3
σ_d	0.0351
ρ_x	0.979
σ_x	0.0003432

The Dynare code reported below implements a first order approximation of the model.

```
var dc, dd, v_c, v_d, x;
```

```
varexo e_c, e_x, e_d;
```

```
parameters DELTA THETA PSI MU_C MU_D RHO_X LAMBDA_DX;
```

```
DELTA=.99;
```

```
PSI=1.5;
```

```
THETA=(1-7.5)/(1-1/PSI);
```

```
MU_C=0.0015;
```

```
MU_D=0.0015;
```

```
RHO_X=.979;
```

```
LAMBDA_DX=3;
```

```
model;
```

```
v_c = DELTA*THETA * exp((-THETA/PSI)*dc(+1) + (THETA-1)*log((1+v_c(+1))*exp(dc(+1))/v_c(+1)) * (1+v_c(+1))*exp(dc(+1)));
```

```
v_d = DELTA*THETA * exp((-THETA/PSI)*dc(+1) + (THETA-1)*log((1+v_c(+1))*exp(dc(+1))/v_c(+1)) * (1+v_d(+1))*exp(dd(+1)));
```

```
dc = MU_C + x(-1) + e_c;
```

```
dd = MU_D + LAMBDA_DX*x(-1) + e_d;
```

```
x = RHO_X * x(-1) + e_x;
```

```
end;
```

```

initval;
v_c=15;
v_d=15;
dc=MU_C;
dd=MU_D;
x=0;
e_c=0;
e_x=0;
e_d=0;
end;

shocks;
var e_c;
stderr .0078;
var e_x;
stderr .0078*.044;
var e_d;
stderr .0078*4.5;
end;

steady;
check;

stoch_simul(order=1, periods=1000, irf=30);

save AssetPricing_simuldata v_d dd dc;

```

Dynare computes policy and transition functions as reported in Table 14. Since these equations are expressed in terms of x_{t-1} , it is possible to express the system in terms of time t state variable using equation (57):

$$v_{c,t} = 49.934 - 8093.043(x_t - \bar{x})$$


$$v_{d,t} = 49.934 - 5610.056(x_t - \bar{x})$$

Equations (55)-(57) complete the description of the solution.

Table 14: Policy and transition functions, CRRA calibration

	Δc	Δd	v_c	v_d	x
Constant	0.0015	0.0015	49.934	49.934	0
$x(-1)$	1	3	-7923.089	-5492.245	0.979
e_c	1	0	0	0	0
e_x	0	0	-8093.043	-5610.056	1
e_d	0	1	0	0	0

It is also interesting to look at the impulse response functions to a shock to x_t . Figure 31 reports the response of the variables of the system to a one standard deviation shock to x . According to the rationale discussed above, both price-dividend and price-consumption ratios fall.

**Figure 31:** Impulse response function to a shock to x_t (CRRA preferences)

11.3.2. Epstein-Zin specification

This section focuses on the more general Epstein-Zin setup. In particular, we allow both risk aversion and IES to be larger than one. Φ and, via equation (50), θ are the only two parameters to differ between Table 13 and Table 15.

Table 15: parameter calibration. The value of θ is calibrated to satisfy equation (50).

Parameter	Calibration
Psi	1.5
θ	-19.5
γ	7.5
δ	0.99
μ_c	0.0015
σ_c	0.0078
μ_d	0.0015
$\lambda_{d,x}$	3
σ_d	0.0351
ρ_x	0.979
σ_x	0.0003432

h

Table 16: Policy and transition functions, CRRA calibration

	Δc	Δd	v_c	v_d	x
Constant	0.0015	0.0015	104.209	104.209	0
x(-1)	1	3	1122.145	7855.017	0.979
e_c	1	0	0	0	0
e_x	0	0	1146.216	8023.511	1
e_d	0	1	0	0	0

By running the Dynare program with the change in the calibration of Φ and θ , the policy functions for v_c and v_d change to:

$$v_{c,t} = 104.209 - 1146.216(x_t - \bar{x})$$

$$v_{d,t} = 104.209 - 8023.511(x_t - \bar{x})$$

Dynare computes policy and transition functions as reported in Table 16.

Therefore, a shock to x_t will now have a positive effect on price-dividend ratios. Separating risk aversion from intertemporal substitution and modelling both as greater than unity produces the effect of changing the sign of the impact of the forecastable part of consumption and dividend growth on price-dividend ratios. The IRF reported in Figure 32 stress this finding.




Figure 32: Impulse response function to a shock to x_t (Epstein-Zin preferences)

11.4. Bayesian estimation

We perform a Bayesian estimation of the general model, using simulated series. The model has three shocks ϵ_c, ϵ_x and ϵ_d and three observables v_d, Δ_c and Δ_d . The priors that we use are reported in Table 17. A slight modification to the program reported in the previous pages enables us to estimate the parameters of the model in Dynare. The following code was been used:

```
var dc, dd, v_c, v_d, x;
varexo e_c, e_x, e_d;
parameters DELTA THETA PSI MU_C MU_D RHO_X LAMBDA_DX;
```

```
DELTA=.99;
PSI=1.5;
THETA=(1-7.5)/(1-1/PSI);
MU_C=0.0015;
MU_D=0.0015;
RHO_X=.979;
LAMBDA_DX=3;
```

```
model; v_c = DELTA*THETA * exp((-THETA/PSI)*dc(+1) + (THETA-1)*log((1+v_c(+1))*exp(dc(+1))
```

```

) * (1+v_c(+1))*exp(dc(+1));
v_d = DELTA*THETA * exp((-THETA/PSI)*dc(+1) + (THETA-1)*log((1+v_c(+1))*exp(dc(+1))/v_c(+1)) * (1+v_d(+1))*exp(dd(+1));
dc = MU_C + x(-1) + e_c;
dd = MU_D + LAMBDA_DX*x(-1) + e_d;
x = RHO_X * x(-1) + e_x;
end;

initval;
dc=MU_C;
dd=MU_D;
v_c=DELTA*exp((1-1/PSI)*dc)/(1-DELTA*exp((1-1/PSI)*dc));
v_d=(DELTA*THETA*exp(-THETA/PSI*dc+dc*(THETA-1)+dd)*(1+v_c)*(THETA-1))/(1/THETA);
x=0;
e_c=0;
e_x=0;
e_d=0;
end;

shocks;
var e_d; stderr .001;
var e_c; stderr .001;
var e_x; stderr .001;
end;

steady;

estimated_params;
DELTA, beta_pdf, 0.98,.005;
THETA,normal_pdf,-19.5, 0.0025;
PSI,normal_pdf,1.6, 0.1;
MU_C,normal_pdf,0.001, 0.001;
MU_D,normal_pdf,0.001, 0.001;
RHO_X,normal_pdf,.98, 0.005;
LAMBDA_DX,normal_pdf,3, 0.05;
stderr e_d,inv_gamma_pdf,.0025, Inf;

```

Table 17: Priors

Parameter	Prior Distribution	Mean	Std.Dev.
Ψ	normal	1.5	0.005
θ	normal	-19.5	0.005
δ	beta	0.99	0.001
μ_c	normal	0.0015	0.005
σ_c	Inv.Gamma	0.0078	inf.
μ_d	normal	0.0015	0.005
$\lambda_{d,x}$	normal	3	0.05
σ_d	Inv.Gamma	0.0351	inf.
ρ_x	normal	0.979	0.05
σ_x	Inv.Gamma	0.0003432	inf.

```
stderr e_x,inv_gamma_pdf,.0003, Inf;
stderr e_c,inv_gamma_pdf,.01, Inf;
end;
```

```
varobs v_d dd dc;
```

```
estimation(datafile=AssetPricing_simuldata,mode_compute=6,mh_replic=1000,mh_jscale=.4,nodia
```


In terms of Dynare's output table 17 displays priors' distributions, while Figure 33 reports a comparison between prior's and posterior's distributions. Since the data generating process is exactly the model that we are estimating, it is not very surprising that the distribution of the posterior has very little variance around the true parameters.

Dynare also produces a table of summary statistics of the estimation. As already seen from the Figures, the mean of the posterior distributions is almost coincidental with the correct values and the estimated confidence interval is very tight around these values. Table 18 reports these numbers.

h

Table 18: Priors and Posteriors

Parameter	Prior Mean	Post.Mean	90% HPD	Interval	Prior Dis.	Post.Std.Dev.
Ψ	1.6	1.5824	1.5428	1.6120	normal	0.1
θ	-19.5	-19.4996	-19.5006	-19.4987	normal	0.0025
δ	0.98	0.9892	0.9884	0.9898	beta	0.005
μ_c	0.001	0.0011	0.0008	0.0015	normal	0.001
σ_c	0.01	0.0079	0.0075	0.0081	Inv.Gamma	inf.
μ_d	0.001	0.002	0.0012	0.0025	normal	0.001
$\lambda_{d,x}$	3	2.9943	2.9736	3.0154	normal	0.05
σ_d	0.003	0.0347	0.0335	0.0360	Inv.Gamma	inf.
ρ_x	0.980	0.9805	0.9787	0.9823	normal	0.005
σ_x	0.000	0.0003	0.0003	0.0004	Inv.Gamma	inf.

**Figure 33:** Posterior)

12. References

Bansal, R., R. Gallant, and G. Tauchen (2007). Rational pessimism, rational exuberance and markets for macro risks. Working Paper .

Bansal, R. and A. Yaron (2004, 08). Risks for the long run: A potential resolution of asset pricing puzzles. *Journal of Finance* 59(4), 1481–1509.

Burnside, C. (1998, March). Solving asset pricing models with gaussian shocks. *Journal of Economic Dynamics and Control* 22(3), 329–340.

Cagan, P. (1956). The monetary dynamics of hyperinflation. In M. Friedman (Ed.), *Studies in the Quantity Theory of Money*. University of Chicago Press.

Cooley, T. F. and E. C. Prescott (1995). Economic growth and business cycles. In T. F. Cooley (Ed.), *Frontiers of Business Cycle Research*. Princeton University Press.

Epstein, L. G. and S. E. Zin (1989, July). Substitution, risk aversion, and the temporal behavior of consumption and asset returns: A theoretical framework. *Econometrica* 57(4), 937–69.

Epstein, L. G. and S. E. Zin (1991, April). Substitution, risk aversion, and the temporal behavior of consumption and asset returns: An empirical analysis. *Journal of Political Economy* 99(2), 263–86.

Hall, R. E. (1988, April). Intertemporal substitution in consumption. *Journal of Political Economy* 96(2), 339–57.

Hansen, L. P., T. J. Sargent, and J. Tallarini, Thomas D (1999, October). Robust permanent income and pricing. *Review of Economic Studies* 66(4), 873–907.

Kim, J. and S. H. Kim (2003, August). Spurious welfare reversals in international business cycle models. *Journal of International Economics* 60(2), 471–500.

Ljungqvist, L. and T. J. Sargent (2004). *Recursive Macroeconomic Theory*, Second edition. Cambridge, Ma.: MIT Press.

Ljungqvist, L. and T. J. Sargent (20XX). Recursive Macroeconomic Theory, Third edition. Cambridge, Ma.: MIT Press.

Ryoo, J. and S. Rosen (2004, February). The engineering labor market. *Journal of Political Economy* 112(S1), S110–S140.

Sargent, T. J. (1977, February). The demand for money during hyperinflations under rational expectations: I. *International Economic Review* 18 (1), 59–82.

Schmitt-Grohe, S. and M. Uribe (2004). Solving asset pricing models with gaussian shocks. *Journal of Economic Dynamics and Control* 28, 755–775.

Slow, A. (1984, May). Occupational choice under uncertainty. *Econometrica* 52(3), 631–45.

A. Appendix List of Dynare Files

This appendix lists the dynare files used in this paper. All files and the data used are contained in the file `examples2019.zip`. In case simulated data was used in the estimation one first needs to run the relevant dynare code to simulate the data.

- `neoclassical_growth_model.mod` - solves the neoclassical growth model and simulates data;
- `neoclassical_growth_model_est.mod` - estimates the neoclassical growth model
- `international_BC.mod` - solves the two country production economy and simulates data
- `international_BC_est.mod` - estimates the two country production economy
- `neoclassical_fiscal_figX.mod` - solves a deterministic growth model and reproduces the graph in chapter 11 of Ljungqvist and Sargent (2004)
- `sargent77.mod` - solves the model in Sargent (1977)
- `sargent77ML.mod`, `sargent77Bayes.mod` - estimate the model in Sargent (1977) using maximum likelihood and Bayesian MCMC methods, respectively
- `hall1.mod` - solves the model in Hall (1988) and simulates data
- `hall1estimateML.mod`, `hall1estimateBayes.mod` - estimate the model in Hall (1988) using maximum likelihood and Bayesian MCMC methods, respectively
- `rosen.mod` - solves the model in Ryoo and Rosen (2004) and simulates data
- `rosenestimateML.mod`, `rosenestimateBayes.mod` - estimate the model in Ryoo and Rosen (2004) using maximum likelihood and Bayesian MCMC methods, respectively
- `BansalYaronML.mod`, `BansalYaronBayes.mod` - estimate the consumption-growth process of Bansal and Yaron (2004) using maximum likelihood and Bayesian MCMC methods, respectively
- `HSTML.mod`, `HSTBayes.mod` - estimate the model in Hansen, Sargent, and Tallarini (1999) using maximum likelihood and Bayesian MCMC methods, respectively
- `AssetPricingApproximation.mod` - solves the model of Bansal and Yaron (2004) and simulates data

- AssetPricingEstimate.mod - estimates the model of Bansal and Yaron (2004)